
Preface

Experienced data analysts know that a successful analysis or meaningful report often requires
more work in acquiring, merging, and transforming data than in specifying the analysis or report
itself. IBM® SPSS® Statistics contains powerful tools for accomplishing and automating these
tasks. While much of this capability is available through the graphical user interface, many of
the most powerful features are available only through command syntax—and you can make the
programming features of its command syntax significantly more powerful by adding the ability
to combine it with a full-featured programming language. This book offers many examples of
the kinds of things that you can accomplish using command syntax by itself and in combination
with other programming language.

For SAS Users

If you have more experience with SAS for data management, see Chapter 32 for comparisons
of the different approaches to handling various types of data management tasks. Quite often,
there is not a simple command-for-command relationship between the two programs, although
each accomplishes the desired end.

Chapter

Overview

This book is divided into several sections:

m Data management using the IBM® SPSS® Statistics command language. Although many of
these tasks can also be performed with the menus and dialog boxes, some very powerful
features are available only with command syntax.

®  Programming with SPSS Statistics and Python. The SPSS Statistics-Python Integration Plug-In
provides the ability to integrate the capabilities of the Python programming language with
SPSS Statistics. One of the major benefits of Python is the ability to add jobwise flow control
to the SPSS Statistics command stream. SPSS Statistics can execute casewise conditional
actions based on criteria that evaluate each case, but jobwise flow control—such as running
different procedures for different variables based on data type or level of measurement, or
determining which procedure to run next based on the results of the last procedure—is much
more difficult. The Python Plug-In makes jobwise flow control much easier to accomplish.
It also provides the ability to operate on output objects—for example, allowing you to
customize pivot tables.

®  Programming with SPSS Statistics and R. The SPSS Statistics-R Integration Plug-In provides
the ability to integrate the capabilities of the R statistical programming language with SPSS
Statistics. This allows you to take advantage of many statistical routines already available
in the R language, plus the ability to write your own routines in R, all from within SPSS
Statistics.

m  Extension commands. Extension commands provide the ability to wrap programs written in
Python or R in SPSS Statistics command syntax. Subcommands and keywords specified in the
command syntax are first validated and then passed as argument parameters to the underlying
Python or R program, which is then responsible for reading any data and generating any
results. Extension commands allow users who are proficient in Python or R to share external
functions with users of SPSS Statistics command syntax.

m SPSS Statistics for SAS programmers. For readers who may be more familiar with the
commands in the SAS system, Chapter 32 provides examples that demonstrate how some
common data management and programming tasks are handled in both SAS and SPSS
Statistics.

Using This Book

This book is intended for use with IBM® SPSS® Statistics release 19 or later. Many examples will
work with earlier versions, but some commands and features are not available in earlier releases.

Most of the examples shown in this book are designed as hands-on exercises that you can
perform yourself. The command files and data files used in the examples are provided in a Zip
file, available from Attp://www.spss.com/spss/data_management _book.htm. All of the sample files
are contained in the examples folder.

B /examples/commands contains SPSS Statistics command syntax files.

B /examples/data contains data files in a variety of formats.

© Copyright SPSS Inc. 1989, 2010 1

B /examples/python contains sample Python files.

B /examples/extensions contains examples of extension commands.

All of the sample command files that contain file access commands assume that you have copied
the examples folder to your local hard drive. For example:

GET FILE='/examples/data/duplicates.sav'.
SORT CASES BY ID_house(A) ID_person(A) int_date(A)
AGGREGATE OUTFILE = '/temp/tempdata.sav'

/BREAK = ID_house ID_person

/DuplicateCount = N.

Many examples, such as the one above, also assume that you have a /temp folder for writing
temporary files.

Python files from /examples/python should be copied to your Python site-packages directory.
The location of this directory depends on your platform. Following are the locations for Python
2.6:

m  For Windows users, the site-packages directory is located in the Lib directory under the
Python 2.6 installation directory—for example, C:\Python26\Lib\site-packages.

m  For Mac OS X 10.4 (Tiger) and 10.5 (Leopard) users, the site-packages directory is located at
/Library/Frameworks/Python.framework/Versions/2.6/lib/python2.6/site-packages.

m  For UNIX users (includes SPSS Statistics for Linux and SPSS Statistics Server for UNIX),
the site-packages directory is located in the /lib/python2.6/ directory under the Python 2.6
installation directory—for example, /usr/local/python26/lib/python2.6/site-packages.

Documentation Resources

The IBM® SPSS® Statistics Core System User’s Guide documents the data management tools
available through the graphical user interface. The material is similar to that available in the
Help system.

The SPSS Statistics Command Syntax Reference, which is installed as a PDF file with the
SPSS Statistics system, is a complete guide to the specifications for each command. The guide
provides many examples illustrating individual commands. It has only a few extended examples
illustrating how commands can be combined to accomplish the kinds of tasks that analysts
frequently encounter. Sections of the SPSS Statistics Command Syntax Reference that are of
particular interest include:

®  The appendix “Defining Complex Files,” which covers the commands specifically intended
for reading common types of complex files.

B The INPUT PROGRAM—END INPUT PROGRAM command, which provides rules for working
with input programs.

All of the command syntax documentation is also available in the Help system. If you type a
command name or place the cursor inside a command in a syntax window and press F1, you will
be taken directly to the help for that command.

Chapter

Best Practices and Efficiency Tips

If you haven’t worked with IBM® SPSS® Statistics command syntax before, you will probably
start with simple jobs that perform a few basic tasks. Since it is easier to develop good habits
while working with small jobs than to try to change bad habits once you move to more complex
situations, you may find the information in this chapter helpful.

Some of the practices suggested in this chapter are particularly useful for large projects
involving thousands of lines of code, many data files, and production jobs run on a regular basis
and/or on multiple data sources.

Working with Command Syntax

You don’t need to be a programmer to write command syntax, but there are a few basic things you
should know. A detailed introduction to command syntax is available in the “Universals” section
in the Command Syntax Reference.

Creating Command Syntax Files

An command file is a simple text file. You can use any text editor to create a command syntax file,
but IBM® SPSS® Statistics provides a number of tools to make your job easier. Most features
available in the graphical user interface have command syntax equivalents, and there are several
ways to reveal this underlying command syntax:

m  Use the Paste button. Make selections from the menus and dialog boxes, and then click the
Paste button instead of the OK button. This will paste the underlying commands into a
command syntax window.

m  Record commands in the log. Sclect Display commands in the log on the Viewer tab in the
Options dialog box (Edit menu > Options), or run the command SET PRINTBACK ON. As you
run analyses, the commands for your dialog box selections will be recorded and displayed in
the log in the Viewer window. You can then copy and paste the commands from the Viewer
into a syntax window or text editor. This setting persists across sessions, so you have to
specify it only once.

m Retrieve commands from the journal file. Most actions that you perform in the graphical user
interface (and all commands that you run from a command syntax window) are automatically
recorded in the journal file in the form of command syntax. The default name of the journal
file is statistics.jnl. The default location varies, depending on your operating system. Both
the name and location of the journal file are displayed on the General tab in the Options
dialog box (Edit > Options).

m Use auto-complete in the Syntax Editor to build command syntax interactively. Starting with
version 17.0, the built-in Syntax Editor contains many tools to help you build and debug
command syntax.

© Copyright SPSS Inc. 1989, 2010 5

Using the Syntax Editor to Build Commands

The Syntax Editor provides assistance in the form of auto-completion of commands,
subcommands, keywords, and keyword values. By default, you are prompted with a
context-sensitive list of available terms. You can display the list on demand by pressing
CTRL4SPACEBAR and you can close the list by pressing the ESC key.

The Auto Complete menu item on the Tools menu toggles the automatic display of the
auto-complete list on or off. You can also enable or disable automatic display of the list from the
Syntax Editor tab in the Options dialog box. Toggling the Auto Complete menu item overrides the
setting on the Options dialog but does not persist across sessions.

Figure 2-1
Auto-complete in Syntax Editor
| i *readexcel.sps - Syntax Editor g@ |

Running Commands

Once you have a set of commands, you can run the commands in a number of ways:

m  Highlight the commands that you want to run in a command syntax window and click the
Run button.

® Invoke one command file from another with the INCLUDE or INSERT command. For more
information, see the topic “Using INSERT with a Master Command Syntax File” on p. 15.

m  Use the Production Facility to create production jobs that can run unattended and even start
unattended (and automatically) using common scheduling software. See the Help system for
more information about the Production Facility.

m  Use IBM® SPSS® Statistics Batch Facility (available only with the server version) to run
command files from a command line and automatically route results to different output
destinations in different formats. See the SPSS Statistics Batch Facility documentation
supplied with the SPSS Statistics server software for more information.

Syntax Rules

®  Commands run from a command syntax window during a typical IBM® SPSS® Statistics
session must follow the interactive command syntax rules.

m  Commands files run via SPSS Statistics Batch Facility or invoked via the INCLUDE command
must follow the batch command syntax rules.

Interactive Rules

The following rules apply to command specifications in interactive mode:

B Each command must start on a new line. Commands can begin in any column of a command
line and continue for as many lines as needed. The exception is the END DATA command,
which must begin in the first column of the first line after the end of data.

m  Each command should end with a period as a command terminator. It is best to omit the
terminator on BEGIN DATA, however, so that inline data are treated as one continuous
specification.

B The command terminator must be the last nonblank character in a command.
®m In the absence of a period as the command terminator, a blank line is interpreted as a

command terminator.

Note: For compatibility with other modes of command execution (including command files run
with INSERT or INCLUDE commands in an interactive session), each line of command syntax
should not exceed 256 bytes.

Batch Rules

The following rules apply to command specifications in batch mode:

B All commands in the command file must begin in column 1. You can use plus (+) or minus
(-) signs in the first column if you want to indent the command specification to make the
command file more readable.

®  [f multiple lines are used for a command, column 1 of each continuation line must be blank.
®  Command terminators are optional.

B A line cannot exceed 256 bytes; any additional characters are truncated.

Protecting the Original Data

The original data file should be protected from modifications that may alter or delete original
variables and/or cases. If the original data are in an external file format (for example, text, Excel,
or database), there is little risk of accidentally overwriting the original data while working in
IBM® SPSS® Statistics. However, if the original data are in SPSS Statistics data files (.sav), there
are many transformation commands that can modify or destroy the data, and it is not difficult to
inadvertently overwrite the contents of a data file in SPSS Statistics format. Overwriting the
original data file may result in a loss of data that cannot be retrieved.

There are several ways in which you can protect the original data, including:
m  Storing a copy in a separate location, such as on a CD, that can’t be overwritten.

m  Using the operating system facilities to change the read-write property of the file to read-only.
If you aren’t familiar with how to do this in the operating system, you can choose Mark File
Read Only from the File menu or use the PERMISSTONS subcommand on the SAVE command.

The ideal situation is then to load the original (protected) data file into SPSS Statistics and do a//
data transformations, recoding, and calculations using SPSS Statistics. The objective is to end
up with one or more command syntax files that start from the original data and produce the
required results without any manual intervention.

Do Not Overwrite Original Variables

It is often necessary to recode or modify original variables, and it is good practice to assign the
modified values to new variables and keep the original variables unchanged. For one thing, this
allows comparison of the initial and modified values to verify that the intended modifications were
carried out correctly. The original values can subsequently be discarded if required.

Example

*These commands overwrite existing variables.

COMPUTE varl=varl*2.

RECODE var2 (1 thru 5 = 1) (6 thru 10 = 2).

*These commands create new variables.

COMPUTE varl_new=varl*2.

RECODE var2 (1 thru 5 = 1) (6 thru 10 = 2) (ELSE=COPY)
/INTO var2_new.

B The difference between the two COMPUTE commands is simply the substitution of a new
variable name on the left side of the equals sign.

B The second RECODE command includes the INTO subcommand, which specifies a new
variable to receive the recoded values of the original variable. ELSE=COPY makes sure that
any values not covered by the specified ranges are preserved.

Using Temporary Transformations

You can use the TEMPORARY command to temporarily transform existing variables for analysis.
The temporary transformations remain in effect through the first command that reads the data (for
example, a statistical procedure), after which the variables revert to their original values.

Example

*temporary.sps.

DATA LIST FREE /varl var2.
BEGIN DATA

12

~N oW
o O\ >

9 10
END DATA.
TEMPORARY .

COMPUTE varl=varl+ 5.
RECODE var2 (1 thru 5=1) (6 thru 10=2).
FREQUENCIES

/VARIABLES=varl var2

/STATISTICS=MEAN STDDEV MIN MAX.
DESCRIPTIVES

/VARIABLES=varl var2

/STATISTICS=MEAN STDDEV MIN MAX.

® The transformed values from the two transformation commands that follow the TEMPORARY
command will be used in the FREQUENCTIES procedure.

B The original data values will be used in the subsequent DESCRIPTIVES procedure, yielding
different results for the same summary statistics.

Under some circumstances, using TEMPORARY will improve the efficiency of a job when
short-lived transformations are appropriate. Ordinarily, the results of transformations are written
to the virtual active file for later use and eventually are merged into the saved IBM® SPSS®
Statistics data file. However, temporary transformations will not be written to disk, assuming that
the command that concludes the temporary state is not otherwise doing this, saving both time and
disk space. (TEMPORARY followed by SAVE, for example, would write the transformations.)

If many temporary variables are created, not writing them to disk could be a noticeable saving
with a large data file. However, some commands require two or more passes of the data. In
this situation, the temporary transformations are recalculated for the second or later passes. If
the transformations are lengthy and complex, the time required for repeated calculation might be
greater than the time saved by not writing the results to disk. Experimentation may be required to
determine which approach is more efficient.

Using Temporary Variables

For transformations that require intermediate variables, use scratch (temporary) variables for

the intermediate values. Any variable name that begins with a pound sign (#) is treated as a
scratch variable that is discarded at the end of the series of transformation commands when IBM®
SPSS® Statistics encounters an EXECUTE command or other command that reads the data (such
as a statistical procedure).

Example

*scratchvar.sps.

DATA LIST FREE / varl.

BEGIN DATA

12345

END DATA.

COMPUTE factor=1.

LOOP #tempvar=1 TO varl.

- COMPUTE factor=factor * #tempvar.
END LOOP.

EXECUTE.

Figure 2-2
Result of loop with scratch variable
%] “Untitled2 [] - Data Editor (=] 9
File Edit “iew Data Transform Analyze Graphs  Utlities  Add-ons  Window Help
1 varl 1
warl factor | war yar war Y
1] 1.00 1.00 L
2 2.00 2.00
3 3.00 B.00
4 4.00 24.00
5 5.00 120.00 ™
4 v \Data View £ variahle View f [<] B

m  The loop structure computes the factorial for each value of var/ and puts the factorial value in
the variable factor.

m  The scratch variable #tempvar is used as an index variable for the loop structure.
m  For each case, the COMPUTE command is run iteratively up to the value of var/.

m  For each iteration, the current value of the variable factor is multiplied by the current loop
iteration number stored in #empvar.

B The EXECUTE command runs the transformation commands, after which the scratch variable
is discarded.

The use of scratch variables doesn’t technically “protect” the original data in any way, but it does
prevent the data file from getting cluttered with extraneous variables. If you need to remove
temporary variables that still exist after reading the data, you can use the DELETE VARIABLES
command to eliminate them.

Use EXECUTE Sparingly

IBM® SPSS® Statistics is designed to work with large data files. Since going through every
case of a large data file takes time, the software is also designed to minimize the number of
times it has to read the data. Statistical and charting procedures always read the data, but most
transformation commands (for example, COMPUTE, RECODE, COUNT, SELECT IF) do not require
a separate data pass.
The default behavior of the graphical user interface, however, is to read the data for
each separate transformation so that you can see the results in the Data Editor immediately.
Consequently, every transformation command generated from the dialog boxes is followed by
an EXECUTE command. So if you create command syntax by pasting from dialog boxes or
copying from the log or journal, your command syntax may contain a large number of superfluous
EXECUTE commands that can significantly increase the processing time for very large data files.
In most cases, you can remove virtually all of the auto-generated EXECUTE commands,

which will speed up processing, particularly for large data files and jobs that contain many
transformation commands.

To turn off the automatic, immediate execution of transformations and the associated pasting of
EXECUTE commands:

» From the menus, choose:
Edit > Options...

» Click the Data tab.

» Select Calculate values before used.

Lag Functions

One notable exception to the above rule is transformation commands that contain lag functions.
In a series of transformation commands without any intervening EXECUTE commands or other
commands that read the data, lag functions are calculated after all other transformations, regardless
of command order. While this might not be a consideration most of the time, it requires special
consideration in the following cases:

B The lag variable is also used in any of the other transformation commands.

B One of the transformations selects a subset of cases and deletes the unselected cases, such as
SELECT IF or SAMPLE.

Example

*lagfunctions.sps.

*create some data.

DATA LIST FREE /varl.

BEGIN DATA

12345

END DATA.

COMPUTE var2=varl.
********************************’
*Lag without intervening EXECUTE.
COMPUTE lagvarl=LAG(varl).
COMPUTE varl=varl*2.

EXECUTE.
********************************.
*Lag with intervening EXECUTE.
COMPUTE lagvar2=LAG(var2) .

EXECUTE.
COMPUTE var2=var2*2.
EXECUTE.
Figure 2-3
Results of lag functions displayed in Data Editor
& “Untitled3 [] - Data Editor (=] 3,
File Edit Wiew Data Transform  Analyze Graphs  Utilities Add-ons  Window  Help
varl | var2 | lageart | lagvar2 | war W
1 2.00 2.00 . . I
2 4.00 4.00 2.00 1.00
3 B.00 B.00 4.00 2.00
4 g.00 8.00 B.00 3.00
5 10000 10000 B.00 4.00
B -
4| » |\ Data View £ variable View f [<] | [»]]

®  Although var! and var2 contain the same data values, lagvarl and lagvar2 are very different
from each other.

®  Without an intervening EXECUTE command, lagvar is based on the transformed values of
varl.

®  With the EXECUTE command between the two transformation commands, the value of lagvar2
is based on the original value of var2.

B Any command that reads the data will have the same effect as the EXECUTE command. For
example, you could substitute the FREQUENCIES command and achieve the same result.

In a similar fashion, if the set of transformations includes a command that selects a subset of cases
and deletes unselected cases (for example, SELECT IF), lags will be computed after the case
selection. You will probably want to avoid case selection criteria based on lag values—unless
you EXECUTE the lags first.

Starting with version 17.0, you can use the SHIFT VALUES command to calculate both lags and
leads. SHIFT VALUES is a procedure that reads the data, resulting in execution of any pending
transformations. This eliminates the potential pitfalls you might encounter with the LAG function.

Using SCASENUM to Select Cases

The value of the system variable SCASENUM is dynamic. If you change the sort order of cases,
the value of SCASENUM for each case changes. If you delete the first case, the case that formerly
had a value of 2 for this system variable now has the value 1. Using the value of $CASENUM with
the SELECT IF command can be a little tricky because SELECT IF deletes each unselected
case, changing the value of SCASENUM for all remaining cases.

For example, a SELECT IF command of the general form:

SELECT IF (SCASENUM > [positive value]).

will delete all cases because regardless of the value specified, the value of $CASENUM for the
current case will never be greater than 1. When the first case is evaluated, it has a value of 1 for
SCASENUM and is therefore deleted because it doesn’t have a value greater than the specified
positive value. The erstwhile second case then becomes the first case, with a value of 1, and is
consequently also deleted, and so on.

The simple solution to this problem is to create a new variable equal to the original value of
SCASENUM. However, command syntax of the form:

COMPUTE CaseNumber=$CASENUM.
SELECT IF (CaseNumber > [positive value]).

will still delete all cases because each case is deleted before the value of the new variable is
computed. The correct solution is to insert an EXECUTE command between COMPUTE and SELECT
IF, as in:

COMPUTE CaseNumber=$CASENUM.
EXECUTE.
SELECT IF (CaseNumber > [positive value]).

MISSING VALUES Command

If you have a series of transformation commands (for example, COMPUTE, IF, RECODE) followed
by a MISSING VALUES command that involves the same variables, you may want to place an
EXECUTE statement before the MISSING VALUES command. This is because the MISSING
VALUES command changes the dictionary before the transformations take place.

Example

IF (x = 0) v = z*2.
MISSING VALUES x (0).

The cases where x = 0 would be considered user-missing on x, and the transformation of y
would not occur. Placing an EXECUTE before MISSING VALUES allows the transformation
to occur before 0 is assigned missing status.

WRITE and XSAVE Commands

In some circumstances, it may be necessary to have an EXECUTE command after a WRITE or an
xSavE command. For more information, see the topic “Using XSAVE in a Loop to Build a
Data File” in Chapter 8 on p. 126.

Using Comments

It is always a good practice to include explanatory comments in your code. You can do this
in several ways:

COMMENT Get summary stats for scale variables.
* An asterisk in the first column also identifies comments.
FREQUENCIES
VARIABLES=income ed reside
/FORMAT=LIMIT(10) /*avoid long frequency tables
/STATISTICS=MEAN /*arithmetic average*/ MEDIAN.
* A macro name like !mymacro in this comment may invoke the macro.
/* A macro name like !mymacro in this comment will not invoke the macro*/.

The first line of a comment can begin with the keyword COMMENT or with an asterisk (*).

Comment text can extend for multiple lines and can contain any characters. The rules for
continuation lines are the same as for other commands. Be sure to terminate a comment
with a period.

m  Use /* and */ to set off a comment within a command.

B The closing */ is optional when the comment is at the end of the line. The command can
continue onto the next line just as if the inserted comment were a blank.

m  To ensure that comments that refer to macros by name don’t accidently invoke those macros,
use the /* [comment text] */ format.

Using SET SEED to Reproduce Random Samples or Values

When doing research involving random numbers—for example, when randomly assigning cases
to experimental treatment groups—you should explicitly set the random number seed value if you
want to be able to reproduce the same results.

The random number generator is used by the SAMPLE command to generate random samples
and is used by many distribution functions (for example, NORMAL, UNIFORM) to generate
distributions of random numbers. The generator begins with a seed, a large integer. Starting with
the same seed, the system will repeatedly produce the same sequence of numbers and will select
the same sample from a given data file. At the start of each session, the seed is set to a value that
may vary or may be fixed, depending on your current settings. The seed value changes each time a
series of transformations contains one or more commands that use the random number generator.

Example

To repeat the same random distribution within a session or in subsequent sessions, use SET SEED
before each series of transformations that use the random number generator to explicitly set
the seed value to a constant value.

*set_seed.sps.

GET FILE = '/examples/data/onevar.sav'.
SET SEED = 123456789.

SAMPLE .1.

LIST.

GET FILE = '/examples/data/onevar.sav'.
SET SEED = 123456789.

SAMPLE .1.

LIST.

Before the first sample is taken the first time, the seed value is explicitly set with SET SEED.

The LIST command causes the data to be read and the random number generator to be
invoked once for each original case. The result is an updated seed value.

m  The second time the data file is opened, SET SEED sets the seed to the same value as before,
resulting in the same sample of cases.

B Both SET SEED commands are required because you aren’t likely to know what the initial
seed value is unless you set it yourself.

Note: This example opens the data file before each SAMPLE command because successive SAMPLE
commands are cumulative within the active dataset.

SET SEED versus SET MTINDEX

There are two random number generators, and SET SEED sets the starting value for only the
default random number generator (SET RNG=MC). If you are using the newer Mersenne Twister
random number generator (SET RNG=MT), the starting value is set with SET MTINDEX.

Divide and Conquer

A time-proven method of winning the battle against programming bugs is to split the tasks into
separate, manageable pieces. It is also easier to navigate around a syntax file of 200—300 lines
than one of 2,000-3,000 lines.

Therefore, it is good practice to break down a program into separate stand-alone files, each
performing a specific task or set of tasks. For example, you could create separate command
syntax files to:

B Prepare and standardize data.

m  Merge data files.

m  Perform tests on data.

m  Report results for different groups (for example, gender, age group, income category).

Using the INSERT command and a master command syntax file that specifies all of the other
command files, you can partition all of these tasks into separate command files.

Using INSERT with a Master Command Syntax File

The INSERT command provides a method for linking multiple syntax files together, making it
possible to reuse blocks of command syntax in different projects by using a “master” command
syntax file that consists primarily of INSERT commands that refer to other command syntax files.

Example

INSERT FILE
INSERT FILE
INSERT FILE
INSERT FILE

"/examples/data/prepare data.sps" CD=YES.
"combine data.sps".

"do tests.sps".

"report groups.sps".

® FEach INSERT command specifies a file that contains command syntax.

m By default, inserted files are read using interactive syntax rules, and each command should
end with a period.

®  The first INSERT command includes the additional specification CD=YES. This changes the
working directory to the directory included in the file specification, making it possible to use
relative (or no) paths on the subsequent INSERT commands.

INSERT versus INCLUDE

INSERT is a newer, more powerful and flexible alternative to INCLUDE. Files included with
INCLUDE must always adhere to batch syntax rules, and command processing stops when the first
error in an included file is encountered. You can effectively duplicate the INCLUDE behavior with
SYNTAX=BATCH and ERROR=STOP on the INSERT command.

Defining Global Settings

In addition to using INSERT to create modular master command syntax files, you can define global
settings that will enable you to use those same command files for different reports and analyses.

Example

You can create a separate command syntax file that contains a set of FILE HANDLE commands
that define file locations and a set of macros that define global variables for client name, output
language, and so on. When you need to change any settings, you change them once in the global
definition file, leaving the bulk of the command syntax files unchanged.

*define_globals.sps.

FILE HANDLE data /NAME='/examples/data'.

FILE HANDLE commands /NAME='/examples/commands'.

FILE HANDLE spssdir /NAME='/program files/spssinc/statistics'.
FILE HANDLE tempdir /NAME='d:/temp'.

DEFINE !enddate()DATE.DMY (1,1,2004) !ENDDEFINE.
DEFINE !olang()English!ENDDEFINE.

DEFINE !client()"ABC Inc"!ENDDEFINE.

DEFINE !title()TITLE !client.!ENDDEFINE.

The first two FILE HANDLE commands define the paths for the data and command syntax
files. You can then use these file handles instead of the full paths in any file specifications.

The third FILE HANDLE command contains the path to the IBM® SPSS® Statistics folder.

This path can be useful if you use any of the command syntax or script files that are installed
with SPSS Statistics.

The last FILE HANDLE command contains the path of a temporary folder. It is very useful
to define a temporary folder path and use it to save any intermediary files created by the
various command syntax files making up the project. The main purpose of this is to avoid
crowding the data folders with useless files, some of which might be very large. Note that
here the temporary folder resides on the D drive. When possible, it is more efficient to keep
the temporary and main folders on different hard drives.

The DEFINE- ! ENDDEFINE structures define a series of macros. This example uses simple
string substitution macros, where the defined strings will be substituted wherever the macro
names appear in subsequent commands during the session.

lenddate contains the end date of the period covered by the data file. This can be useful to
calculate ages or other duration variables as well as to add footnotes to tables or graphs.

lolang specifies the output language.
Iclient contains the client’s name. This can be used in titles of tables or graphs.

Ititle specifies a TITLE command, using the value of the macro /client as the title text.

The master command syntax file might then look something like this:

INSERT FILE

"/examples/commands/define_globals.sps".

ltitle.

INSERT FILE
INSERT FILE
INSERT FILE

"/data/prepare data.sps".
" /commands/combine data.sps".
"/commands/do tests.sps".

INCLUDE FILE = "/commands/report groups.sps'.

®  The first INSERT runs the command syntax file that defines all of the global settings. This

needs to be run before any commands that invoke the macros defined in that file.

B ! title will print the client’s name at the top of each page of output.

B "data" and "commands" in the remaining INSERT commands will be expanded to
"/examples/data" and "/examples/commands", respectively.

Note: Using absolute paths or file handles that represent those paths is the most reliable way to
make sure that SPSS Statistics finds the necessary files. Relative paths may not work as you might
expect, since they refer to the current working directory, which can change frequently. You can
also use the CD command or the cD keyword on the INSERT command to change the working
directory.

Getting Data into IBM SPSS Statistics

Before you can work with data in IBM® SPSS® Statistics, you need some data to work with.
There are several ways to get data into the application:

®  Open a data file that has already been saved in SPSS Statistics format.
®  Enter data manually in the Data Editor.

m Read a data file from another source, such as a database, text data file, spreadsheet, SAS, or
Stata.

Opening SPSS Statistics data files is simple, and manually entering data in the Data Editor is not
likely to be your first choice, particularly if you have a large amount of data. This chapter focuses
on how to read data files created and saved in other applications and formats.

Getting Data from Databases

IBM® SPSS® Statistics relies primarily on ODBC (open database connectivity) to read data from
databases. ODBC is an open standard with versions available on many platforms, including
Windows, UNIX, Linux, and Macintosh.

Installing Database Drivers

You can read data from any database format for which you have a database driver. In local analysis
mode, the necessary drivers must be installed on your local computer. In distributed analysis mode
(available with the Server version), the drivers must be installed on the remote server.

ODBC database drivers are available for a wide variety of database formats, including:

Access
Btrieve
DB2
dBASE
Excel
FoxPro
Informix
Oracle
Paradox
Progress
SQL Base
SQL Server
Sybase

© Copyright SPSS Inc. 1989, 2010 19

For Windows and Linux operating systems, many of these drivers can be installed by installing
the Data Access Pack. You can install the Data Access Pack from the AutoPlay menu on the
installation DVD.

Before you can use the installed database drivers, you may also need to configure the drivers.
For the Data Access Pack, installation instructions and information on configuring data sources
are located in the Installation Instructions folder on the installation DVD.

OLE DB
Starting with release 14.0, some support for OLE DB data sources is provided.

To access OLE DB data sources (available only on Microsoft Windows operating systems),
you must have the following items installed:

m  NET framework. To obtain the most recent version of the .NET framework, go to
http://www.microsoft.com/net.

m [BM® SPSS® Data Collection Survey Reporter Developer Kit. For information on obtaining
a compatible version of SPSS Survey Reporter Developer Kit, go to support.spss.com
(http://support.spss.com).

The following limitations apply to OLE DB data sources:
m  Table joins are not available for OLE DB data sources. You can read only one table at a time.

B You can add OLE DB data sources only in local analysis mode. To add OLE DB data sources
in distributed analysis mode on a Windows server, consult your system administrator.

m In distributed analysis mode (available with IBM® SPSS® Statistics Server), OLE DB data
sources are available only on Windows servers, and both .NET and SPSS Survey Reporter
Developer Kit must be installed on the server.

Database Wizard

It’s probably a good idea to use the Database Wizard (File > Open Database) the first time you
retrieve data from a database source. At the last step of the wizard, you can paste the equivalent
commands into a command syntax window. Although the SQL generated by the wizard tends
to be overly verbose, it also generates the CONNECT string, which you might never figure out
without the wizard.

Reading a Single Database Table

IBM® SPSS® Statistics reads data from databases by reading database tables. You can read
information from a single table or merge data from multiple tables in the same database. A single
database table has basically the same two-dimensional structure as a data file in SPSS Statistics
format: records are cases and fields are variables. So, reading a single table can be very simple.


http://www.microsoft.com/net
http://support.spss.com
http://support.spss.com
http://support.spss.com
http://support.spss.com

Example

This example reads a single table from an Access database. It reads all records and fields in
the table.

*accessl.sps.
GET DATA /TYPE=ODBC /CONNECT=
'DSN=Microsoft Access;DBQ=c:\examples\data\dm_demo.mdb; '+
' DriverId=25;FIL=MS Access;MaxBufferSize=2048;PageTimeout=5;"
/SQL = 'SELECT * FROM CombinedTable'.
EXECUTE.

B The GET DATA command is used to read the database.

B  TYPE=0DBC indicates that an ODBC driver will be used to read the data. This is required for
reading data from any database, and it can also be used for other data sources with ODBC
drivers, such as Excel workbooks. For more information, see the topic “Reading Multiple
Worksheets” on p. 30.

B CONNECT identifies the data source. For this example, the CONNECT string was copied from
the command syntax generated by the Database Wizard. The entire string must be enclosed in
single or double quotes. In this example, we have split the long string onto two lines using
a plus sign (+) to combine the two strings.

B  The sQL subcommand can contain any SQL statements supported by the database format.
Each line must be enclosed in single or double quotes.

B SELECT * FROM CombinedTable reads all of the fields (columns) and all records (rows)
from the table named CombinedTable in the database.

B Any field names that are not valid variable names are automatically converted to valid variable
names, and the original field names are used as variable labels. In this database table, many of
the field names contain spaces, which are removed in the variable names.

Figure 3-1
Database field names converted to valid variable names

%] *Untitled2 [] - Data Editor =
File Edit Yew Data Transform Analvze Graphs  Utlities  Add-ons  Window Help
Name | Type | Width | Decimals | Label &

111D Murneric 11 0

2|Age Murmeric 8 2

3| MaritalStatus MNurneric 8 2 Marital Status

4|Income MNurneric 8 2

5|IncomeCategory |Mumeric 8 2 Income Categary

B|Car MNureric 8 2

7| CarCategory MNumeric g 2 Car Category

8| Education Murneric g 2

9|Employ Mumeric g 2

A0 Pmtion AT o A v
/v |, Data view }Variable View / < >
Example

Now we’ll read the same database table—except this time, we’ll read only a subset of fields
and records.

*access2.sps.

GET DATA /TYPE=ODBC /CONNECT=
'DSN=MS Access Database;DBQ=C:\examples\data\dm_demo.mdb; '+
'DriverId=25;FIL=MS Access;MaxBufferSize=2048;PageTimeout=5;"
/SQL =
'SELECT Age, Education, [Income Categoryl]'
' FROM CombinedTable'
' WHERE ([Marital Status] <> 1 AND Internet = 1 )'.

EXECUTE.

m The SeLECT clause explicitly specifies only three fields from the file; so, the active dataset
will contain only three variables.

® The WHERE clause will select only records where the value of the Marital Status field is not 1
and the value of the Internet field is 1. In this example, that means only unmarried people
who have Internet service will be included.

Two additional details in this example are worth noting:

m  The field names Income Category and Marital Status are enclosed in brackets. Since these
field names contain spaces, they must be enclosed in brackets or quotes. Since single quotes
are already being used to enclose each line of the SQL statement, the alternative to brackets
here would be double quotes.

m  We’ve put the FROM and WHERE clauses on separate lines to make the code easier to read,;
however, in order for this command to be read properly, each of those lines also has a blank
space between the starting single quote and the first word on the line. When the command
is processed, all of the lines of the SQL statement are merged together in a very literal
fashion. Without the space before WHERE, the program would attempt to read a table named
CombinedTableWhere, and an error would result. As a general rule, you should probably
insert a blank space between the quotation mark and the first word of each continuation line.

Reading Multiple Tables

You can combine data from two or more database tables by “joining” the tables. The active
dataset can be constructed from more than two tables, but each “join” defines a relationship
between only two of those tables:

® Inner join. Records in the two tables with matching values for one or more specified fields
are included. For example, a unique ID value may be used in each table, and records with
matching ID values are combined. Any records without matching identifier values in the
other table are omitted.

m  Left outer join. All records from the first table are included regardless of the criteria used to
match records.

m  Right outer join. Essentially the opposite of a left outer join. So, the appropriate one to use is
basically a matter of the order in which the tables are specified in the SQL SELECT clause.

Example

In the previous two examples, all of the data resided in a single database table. But what if the
data were divided between two tables? This example merges data from two different tables: one
containing demographic information for survey respondents and one containing survey responses.

*access_multtablesl.sps.
GET DATA /TYPE=0ODBC /CONNECT=
'DSN=MS Access Database;DBQ=C:\examples\data\dm_demo.mdb; '+
'DriverId=25;FIL=MS Access;MaxBufferSize=2048;PageTimeout=5;"
/SQL =
'SELECT * FROM DemographicInformation, SurveyResponses'
' WHERE DemographicInformation.ID=SurveyResponses.ID'.
EXECUTE.

m  The SELECT clause specifies all fields from both tables.

B The WHERE clause matches records from the two tables based on the value of the ID field in
both tables. Any records in either table without matching /D values in the other table are
excluded.

®  The result is an inner join in which only records with matching /D values in both tables
are included in the active dataset.

Example

In addition to one-to-one matching, as in the previous inner join example, you can also merge
tables with a one-to-many matching scheme. For example, you could match a table in which there
are only a few records representing data values and associated descriptive labels with values in a
table containing hundreds or thousands of records representing survey respondents.

In this example, we read data from an SQL Server database, using an outer join to avoid
omitting records in the larger table that don’t have matching identifier values in the smaller table.

*sglserver_outer_join.sps.
GET DATA /TYPE=0ODBC
/CONNECT= 'DSN=SQLServer;UID=;APP=SPSS Statistics;"'
'WSID=USERLAP; Network=DBMSSOCN; Trusted_Connection=Yes'
/SQL =
'SELECT SurveyResponses.ID, SurveyResponses.Internet, '
' [Value Labels].[Internet Label]'
' FROM SurveyResponses LEFT OUTER JOIN [Value Labels]'
' ON SurveyResponses.Internet'
' = [Value Labels].[Internet Value]'.

Figure 3-2
SQL Server tables to be merged with outer join

Figure 3-3
Active dataset in IBM SPSS Statistics

B FROM SurveyResponses LEFT OUTER JOIN [Value Labels] will include all records

from the table SurveyResponses even if there are no records in the Value Labels table that
meet the matching criteria.

ON SurveyResponses.Internet = [Value Labels].[Internet Value] matches
records based on the value of the field Infernet in the table SurveyResponses and the value of
the field Internet Value in the table Value Labels.

The resulting active dataset has an Internet Label value of No for all cases with a value of 0
for Internet and Yes for all cases with a value of 1 for Internet.

Since the left outer join includes all records from SurveyResponses, there are cases in the
active dataset with values of 8 or 9 for Internet and no value (a blank string) for Internet Label,
since the values of 8 and 9 do not occur in the Internet Value field in the table Value Labels.

Reading IBM SPSS Statistics Data Files with SQL Statements

You can select subsets of variables when you read IBM® SPSS® Statistics data files with the GET
command, and you can select subsets of cases with SELECT IF. You can also use standard SQL
statements to read subsets of variables and cases using the SPSS Statistics Data File Driver.

The SPSS Statistics data file driver allows you to read SPSS Statistics (.sav) data files in
applications that support Open Database Connectivity (ODBC) or Java Database Connectivity
(JDBC). SPSS Statistics itself supports ODBC in the Database Wizard, providing you with the
ability to leverage the Structured Query Language (SQL) when reading .sav data files in SPSS
Statistics.

There are three flavors of the SPSS Statistics data file driver, all of which are available for
Windows, UNIX, and Linux:

m  Standalone driver. The standalone driver provides ODBC support without requiring
installation of additional components. After the standalone driver is installed, you can
immediately set up an ODBC data source and use it to read .sav files.

m  Service driver. The service driver provides both ODBC and JDBC support. The service driver
handles data requests from the service client driver, which may be installed on the same
computer or on one or more remote computers. Thus you can configure one service driver
that may be used by many clients. If you put your data files on the same computer on which
the service driver is installed, the service driver can reduce network traffic because all the
queries occur on the server. Only the resulting cases are sent to the service client. If the
server has a faster processor or more RAM compared to service client machines, there may
also be performance improvements.

m  Service client driver. The service client driver provides an interface between the client
application that needs to read the .sav data file and the service driver that handles the request
for the data. Unlike the standalone driver, it supports both ODBC and JDBC. The operating
system of the service client driver does not need to match the operating system of the service
driver. For example, you can install the service driver on a UNIX machine and the service
client driver on a Windows machine.

Using the standalone and service client drivers is similar to connecting to a database with any
other ODBC or JDBC driver. After configuring the driver, creating data sources, and connecting
to the SPSS Statistics data file, you will see that the data file is represented as a collection of
tables. In other words, the data file looks like a database source.

Installing the IBM SPSS Statistics Data File Driver

You can download and install the IBM® SPSS® Statistics data file driver from
http://www.spss.com/drivers/. The SPSS Statistics Data File Driver Guide, available from the
same location, contains information on installing and configuring the driver.


http://www.spss.com/drivers/

Using the Standalone Driver

This example uses the ODBC standalone driver to select a subset of variables and cases when
reading a data file in IBM® SPSS® Statistics format into SPSS Statistics. For an example of how
to use the driver to read SPSS Statistics data files into other applications, see “Example: Using the
Standalone Driver with Excel” in Chapter 9.

*sav_odbc.sps.

GET DATA
/TYPE=0ODBC
/CONNECT=
"DRIVER=IBM SPSS Statistics 19 Data File Driver - Standalone;"
"SDSN=SAVDB; "
"HST=C:\Program Files\IBM\SPSS\StatisticsDataFileDriver\19"

"\Standalone\cfg\oadm.ini;"

"PRT=StatisticsSAVDriverStandalone;"
"CP_CONNECT_STRING=C:\examples\data\demo.sav;"
"CP_UserMissingIsNull=0"
/SQL="SELECT age, marital, inccat, gender FROM demo.Cases "
"WHERE (age > 40 AND gender = 'm')".

CACHE.

EXECUTE.

APPLY DICTIONARY FROM '/examples/data/demo.sav'.

®  DRIVER. Instead of specifying a DSN (data source name), the CONNECT statement specifies
the driver name. You could define DSNs for each SPSS Statistics data file that you want to
access with the ODBC driver (using the ODBC Data Source Administrator on Windows),
but specifying the driver and all other parameters on the CONNECT statement makes it easier
to reuse and modify the same basic syntax for different data files. The driver name is
always SPSS Statistics <version> Data File Driver - Standalone, where
<version> is the product version number.

SDSN. This is always set to SAVDB.

HST. This specifies the location of the oadm.ini file. It is located in the c¢fg sub-directory
of the driver installation directory.

m  PRT. This is always set to StatisticsSAVDriverStandalone.

m  CP_CONNECT_STRING. The full path and name of the SPSS Statistics data file. This path
cannot contain an equals sign (=) or semicolon (;).

m  CP_UserMissinglsNull. This specifies the treatment of user-defined missing values. If it is set
to 0, user-defined missing values are read as valid values. If it is set to 1, user-defined missing
values are set to system-missing for numeric variables and blank for string variables. In this
example, the user-defined missing values will be read as valid values and then the original
user-missing definitions will be reapplied with APPLY DICTIONARY.

m  SQOL. The SQL subcommand uses standard SQL syntax to specify the variables (fields) to
include, the name of the database table, and the case (record) selection rules.

m  SELECT specifies the subset of variables (fields) to read. In this example, the variables age,
marital, inccat, and gender.

m  FROM specifies the database table to read. The prefix is the name of the SPSS Statistics data
file. The Cases table contains the case data values.

m  WHERE specifies the criteria for selecting cases (records). In this example, males over 40
years of age.

m  APPLY DICTIONARY applies the dictionary information (variable labels, value labels, missing
value definitions, and so forth) from the original SPSS Statistics data file. When you use
GET DATA /TYPE=O0DBC to read SPSS Statistics data files, the dictionary information is not
included, but this is easily restored with APPLY DICTIONARY.

Reading Excel Files

You can read individual Excel worksheets and multiple worksheets in the same Excel workbook.
The basic mechanics of reading Excel files are relatively straightforward—rows are read as cases
and columns are read as variables. However, reading a typical Excel spreadsheet—where the data
may not start in row 1, column 1—requires a little extra work, and reading multiple worksheets
requires treating the Excel workbook as a database. In both instances, we can use the GET DATA
command to read the data.

Reading a “Typical” Worksheet

When reading an individual worksheet, IBM® SPSS® Statistics reads a rectangular area of the
worksheet, and everything in that area must be data related. The first row of the area may or may
not contain variable names (depending on your specifications); the remainder of the area must
contain the data to be read. A typical worksheet, however, may also contain titles and other
information that may not be appropriate for data in SPSS Statistics and may even cause the data to
be read incorrectly if you don’t explicitly specify the range of cells to read.

Example
Figure 3-4
Typical Excel worksheet

To read this spreadsheet without the title row or total row and column:

*readexcel.sps.
GET DATA

/TYPE=XLS
/FILE="'/examples/data/sales.xls'
/SHEET=NAME 'Gross Revenue'
/CELLRANGE=RANGE 'A2:I15"
/READNAMES=0on

The TYPE subcommand identifies the file type as Excel 95 or later. For earlier versions, use
GET TRANSLATE. For Excel 2007 or later, user GET DATA /TYPE=XLSX (or XLSM).

The SHEET subcommand identifies which worksheet of the workbook to read. Instead of the
NaME keyword, you could use the INDEX keyword and an integer value indicating the sheet
location in the workbook. Without this subcommand, the first worksheet is read.

The CELLRANGE subcommand indicates that the data should be read starting at column A4,
row 2, and read through column 7, row 15.

The READNAMES subcommand indicates that the first row of the specified range contains
column labels to be used as variable names.

Figure 3-5
Excel worksheet read into IBM SPSS Statistics

m  The Excel column label Store Number is automatically converted to the variable name
StoreNumber, since variable names cannot contain spaces. The original column label is
retained as the variable label.

m  The original data type from Excel is preserved whenever possible, but since data type is
determined at the individual cell level in Excel and at the column (variable) level in SPSS
Statistics, this isn’t always possible.

®  When mixed data types exist in the same column, the variable is assigned the string data type;
s0, the variable Toys in this example is assigned the string data type.

READNAMES Subcommand

The READNAMES subcommand treats the first row of the spreadsheet or specified range as either
variable names (ON) or data (OFF). This subcommand will always affect the way the Excel
spreadsheet is read, even when it isn’t specified, since the default setting is ON.

®  With READNAMES=0N (or in the absence of this subcommand), if the first row contains data
instead of column headings, IBM® SPSS® Statistics will attempt to read the cells in that row
as variable names instead of as data—alphanumeric values will be used to create variable
names, numeric values will be ignored, and default variable names will be assigned.

m  With READNAMES=0FF, if the first row does, in fact, contain column headings or other
alphanumeric text, then those column headings will be read as data values, and all of the
variables will be assigned the string data type.

Reading Multiple Worksheets

An Excel file (workbook) can contain multiple worksheets, and you can read multiple worksheets
from the same workbook by treating the Excel file as a database. This requires an ODBC driver
for Excel.

Figure 3-6
Multiple worksheets in same workbook

When reading multiple worksheets, you lose some of the flexibility available for reading
individual worksheets:

B You cannot specify cell ranges.

m  The first non-empty row of each worksheet should contain column labels that will be used
as variable names.

®  Only basic data types—string and numeric—are preserved, and string variables may be set to
an arbitrarily long width.

Example

In this example, the first worksheet contains information about store location, and the second and
third contain information for different departments. All three contain a column, Store Number,
that uniquely identifies each store, so, the information in the three sheets can be merged correctly
regardless of the order in which the stores are listed on each worksheet.

*readexcel2.sps.
GET DATA
/TYPE=0ODBC
/CONNECT=
'DSN=Excel Files;DBQ=c:\examples\data\sales.xls;' +
'DriverId=790;MaxBufferSize=2048; PageTimeout=5; "'
/SQL =

'SELECT Location$.[Store Number], State, Region, City,'

' Power, Hand, Accessories,'

' Tires, Batteries, Gizmos, Dohickeys'

' FROM [Location$], [Tools$], [AutoS]'

' WHERE [Tools$].[Store Number]=[Location$].[Store Number]'
' AND [Auto$].[Store Number]=[Location$].[Store Number]'.

If these commands look like random characters scattered on the page to you, try using the
Database Wizard (File > Open Database) and, in the last step, paste the commands into a
syntax window.

Even if you are familiar with SQL statements, you may want to use the Database Wizard the
first time to generate the proper CONNECT string.

The SELECT statement specifies the columns to read from each worksheet, as identified by
the column headings. Since all three worksheets have a column labeled Store Number, the
specific worksheet from which to read this column is also included.

If the column headings can’t be used as variable names, you can either let IBM® SPSS®
Statistics automatically create valid variable names or use the As keyword followed by a
valid variable name. In this example, Store Number is not a valid variable name; so, a
variable name of StoreNumber is automatically created, and the original column heading is
used as the variable label.

The FROM clause identifies the worksheets to read.

The WHERE clause indicates that the data should be merged by matching the values of the
column Store Number in the three worksheets.

Figure 3-7

Merged worksheets in IBM SPSS Statistics

Reading Text Data Files

A text data file is simply a text file that contains data. Text data files fall into two broad categories:

m  Simple text data files, in which all variables are recorded in the same order for all cases, and
all cases contain the same variables. This is basically how all data files appear once they are
read into IBM® SPSS® Statistics.

m  Complex text data files, including files in which the order of variables may vary between cases
and hierarchical or nested data files in which some records contain variables with values that
apply to one or more cases contained on subsequent records that contain a different set of
variables (for example, city, state, and street address on one record and name, age, and gender
of each household member on subsequent records).

Text data files can be further subdivided into two more categories:

m Delimited. Spaces, commas, tabs, or other characters are used to separate variables. The
variables are recorded in the same order for each case but not necessarily in the same column
locations. This is also referred to as freefield format. Some applications export text data in
comma-separated values (CSV) format; this is a delimited format.

m  Fixed width. Each variable is recorded in the same column location on the same line (record)
for each case in the data file. No delimiter is required between values. In fact, in many text
data files generated by computer programs, data values may appear to run together without
even spaces separating them. The column location determines which variable is being read.

Complex data files are typically also fixed-width format data files.

Simple Text Data Files

In most cases, the Text Wizard (File > Read Text Data) provides all of the functionality that you
need to read simple text data files. You can preview the original text data file and resulting IBM®
SPSS® Statistics data file as you make your choices in the wizard, and you can paste the command
syntax equivalent of your choices into a command syntax window at the last step.

Two commands are available for reading text data files: GET DATA and DATA LIST. In many
cases, they provide the same functionality, and the choice of one versus the other is a matter of
personal preference. In some instances, however, you may need to take advantage of features in
one command that aren’t available in the other.

GET DATA

Use GET DATA instead of DATA LIST if:
m The file is in CSV format.
m  The text data file is very large.

DATA LIST

Use DATA LIST instead of GET DATA if:

m  The text data is “inline” data contained in a command syntax file using BEGIN DATA—-END
DATA.

m The file has a complex structure, such as a mixed or hierarchical structure. For more
information, see the topic “Reading Complex Text Data Files” on p. 41.

B You want to use the TO keyword to define a large number of sequential variable names (for
example, varl TO varl1000).

B You need to specify the encoding of the text file. For more information, see the topic “Code
Page and Unicode Data Sources” on p. 52.

Many examples in other chapters use DATA LIST to define sample data simply because it supports
the use of inline data contained in the command syntax file rather than in an external data file,
making the examples self-contained and requiring no additional files to work.

Delimited Text Data

In a simple delimited (or “freefield”) text data file, the absolute position of each variable isn’t
important; only the relative position matters. Variables should be recorded in the same order for
each case, but the actual column locations aren’t relevant. More than one case can appear on the
same record, and some records can span multiple records, while others do not.

Example

One of the advantages of delimited text data files is that they don’t require a great deal of structure.
The sample data file, simple_delimited.txt, looks like this:

8
4

1 122122 £2921212
00 532145128 m17 11
1

oS
NN

The bATA LIST command to read the data file is:

*simple_delimited.sps.
DATA LIST FREE

FILE = '/examples/data/simple_delimited.txt'
/id (F3) sex (Al) age (F2) opinionl TO opinion5 (5F).
EXECUTE.

B FREE indicates that the text data file is a delimited file, in which only the order of variables
matters. By default, commas and spaces are read as delimiters between data values. In this
example, all of the data values are separated by spaces.

m  Eight variables are defined, so after reading eight values, the next value is read as the first
variable for the next case, even if it’s on the same line. If the end of a record is reached before
eight values have been read for the current case, the first value on the next line is read as the
next value for the current case. In this example, four cases are contained on three records.

m [f all of the variables were simple numeric variables, you wouldn’t need to specify the format
for any of them, but if there are any variables for which you need to specify the format, any
preceding variables also need format specifications. Since you need to specify a string format
for sex, you also need to specify a format for id.

® In this example, you don’t need to specify formats for any of the numeric variables that appear
after the string variable, but the default numeric format is F8.2, which means that values are
displayed with two decimals even if the actual values are integers. (F2) specifies an integer
with a maximum of two digits, and (5F) specifies five integers, each containing a single digit.

The “defined format for all preceding variables” rule can be quite cumbersome, particularly if you
have a large number of simple numeric variables interspersed with a few string variables or other
variables that require format specifications. You can use a shortcut to get around this rule:

DATA LIST FREE
FILE = '/examples/data/simple_delimited.txt'
/id * sex (Al) age opinionl TO opinion5.

The asterisk indicates that all preceding variables should be read in the default numeric format
(F8.2). In this example, it doesn’t save much over simply defining a format for the first variable,
but if sex were the last variable instead of the second, it could be useful.

Example

One of the drawbacks of DATA LIST FREE is that if a single value for a single case is accidently
missed in data entry, all subsequent cases will be read incorrectly, since values are read
sequentially from the beginning of the file to the end regardless of what line each value is recorded
on. For delimited files in which each case is recorded on a separate line, you can use DATA LIST
LIsT, which will limit problems caused by this type of data entry error to the current case.

The data file, delimited_list.txt, contains one case that has only seven values recorded, whereas
all of the others have eight:

00l m28 12212
002 £ 2921212
003 £ 45 3 2 4 5

128 m 17 111 9 4

The DATA LIST command to read the file is:

*delimited_list.sps.
DATA LIST LIST
FILE='/examples/data/delimited_list.txt'
/id(F3) sex (Al) age opinionl TO opinion5 (6F1).
EXECUTE.

Figure 3-8
Text data file read with DATA LIST LIST

m  FEight variables are defined, so eight values are expected on each line.

m  The third case, however, has only seven values recorded. The first seven values are read as the
values for the first seven defined variables. The eighth variable is assigned the system-missing
value.

You don’t know which variable for the third case is actually missing. In this example, it could be
any variable after the second variable (since that’s the only string variable, and an appropriate
string value was read), making all of the remaining values for that case suspect; so, a warning
message is issued whenever a case doesn’t contain enough data values:

>Warning # 1116

>Under LIST input, insufficient data were contained on one record to
>fulfill the variable list.

>Remaining numeric variables have been set to the system-missing
>value and string variables have been set to blanks.

>Command line: 6 Current case: 3 Current splitfile group: 1

CSV Delimited Text Files

A CSV file uses commas to separate data values and encloses values that include commas in
quotation marks. Many applications export text data in this format. To read CSV files correctly,
you need to use the GET DATA command.

Example

The file CSV _file.csv was exported from Microsoft Excel:

ID,Name, Gender,Date Hired, Department
1, "Foster, Chantal",f,10/29/1998,1
"Healy, Jonathan",m,3/1/1992,3
"Walter, Wendy",£f,1/23/1995,2

2
3
4,"0Oliver, Kendall",f,10/28/2003,2

This data file contains variable descriptions on the first line and a combination of string and
numeric data values for each case on subsequent lines, including string values that contain
commas. The GET DATA command syntax to read this file is:

*delimited_csv.sps.
GET DATA /TYPE = TXT

/FILE = '/examples/data/CSV_file.csv'
/DELIMITERS = ", "

/QUALIFIER = '"'

/ARRANGEMENT = DELIMITED

/FIRSTCASE = 2

/VARIABLES = ID F3 Name Al5 Gender Al

Date_Hired ADATE10 Department F1.

B DELIMITERS = ", " specifies the comma as the delimiter between values.

B QUALIFIER = '"' specifies that values that contain commas are enclosed in double quotes
so that the embedded commas won’t be interpreted as delimiters.

B FIRSTCASE = 2 skips the top line that contains the variable descriptions; otherwise, this
line would be read as the first case.

B ADATE1O0 specifies that the variable Date Hired is a date variable of the general format
mm/dd/yyyy. For more information, see the topic “Reading Different Types of Text Data”
on p. 40.

Note: The command syntax in this example was adapted from the command syntax generated by
the Text Wizard (File > Read Text Data), which automatically generated valid variable names from
the information on the first line of the data file.

Fixed-Width Text Data

In a fixed-width data file, variables start and end in the same column locations for each case. No
delimiters are required between values, and there is often no space between the end of one value
and the start of the next. For fixed-width data files, the command that reads the data file (GET
DATA or DATA LIST) contains information on the column location and/or width of each variable.

Example

In the simplest type of fixed-width text data file, each case is contained on a single line (record) in
the file. In this example, the text data file simple_fixed.txt looks like this:

001 m 28 12212
002 £ 29 21212
003 £ 45 32145
128 m 17 11194

Using DATA LIST, the command syntax to read the file is:

*simple_fixed.sps.
DATA LIST FIXED

FILE='/examples/data/simple_fixed.txt'

/id 1-3 sex 5 (A) age 7-8 opinionl TO opinionb5 10-14.
EXECUTE.

®  The keyword FIXED is included in this example, but since it is the default format, it can
be omitted.

m  The forward slash before the variable id separates the variable definitions from the rest of the
command specifications (unlike other commands where subcommands are separated by
forward slashes). The forward slash actually denotes the start of each record that will be read,
but in this case there is only one record per case.

m  The variable id is located in columns 1 through 3. Since no format is specified, the standard
numeric format is assumed.

®  The variable sex is found in column 5. The format (a) indicates that this is a string variable,
with values that contain something other than numbers.

The numeric variable age is in columns 7 and 8.

opinionl TO opinion5 10-14 defines five numeric variables, with each variable
occupying a single column: opinionl! in column 10, opinion2 in column 11, and so on.

You could define the same data file using variable width instead of column locations:

*simple_fixed_alt.sps.

DATA LIST FIXED
FILE='/examples/data/simple_fixed.txt'
/id (F3, 1X) sex (Al, 1X) age (F2, 1X)
opinionl TO opinion5 (5F1).

EXECUTE.

B id (F3, 1X) indicates that the variable id is in the first three column positions, and the next
column position (column 4) should be skipped.

m  FEach variable is assumed to start in the next sequential column position; so, sex is read from

column 5.
Figure 3-9
Fixed-width text data file displayed in Data Editor
%] *Untitledd [] - Data Editor (=JIo)E=
File Edit Yew Data Transform Analvze Graphs  Utilities  Add-ons  Window  Help
8:id Yisible:
id | sex | age | opiniont | opinion2 | opinion3 | opiniond | opinions | ™S
1 1im 28 1 2 2 1 2
2 20f 28 2 1 2 1 2
3 3f 45 3 2 4 5 .
4] 128|m 17 1 1 1 9 4
5
]
7 v
4/ » |\ Data View A variahle View f < >
Example

Reading the same file with GET DATA, the command syntax would be:

*simple_ fixed_getdata.sps.

GET DATA /TYPE = TXT

/FILE = '/examples/data/simple_fixed.txt'
/ARRANGEMENT = FIXED

/VARIABLES =/1 id 0-2 F3 sex 4-4 Al age 6-7 F2
opinionl 9-9 F opinion2 10-10 F opinion3 11-11 F
opinion4 12-12 F opinionb5 13-13 F.

B The first column is column O (in contrast to DATA LIST, in which the first column is column
1).
m  There is no default data type. You must explicitly specify the data type for all variables.

B You must specify both a start and an end column position for each variable, even if the
variable occupies only a single column (for example, sex 4-4).

m  All variables must be explicitly specified; you cannot use the keyword TO to define a range of
variables.

Reading Selected Portions of a Fixed-Width File

With fixed-format text data files, you can read all or part of each record and/or skip entire records.

Example

In this example, each case takes two lines (records), and the first line of the file should be skipped
because it doesn’t contain data. The data file, skip_first fixed.txt, looks like this:

Employee age, department, and salary information
John Smith

26 2 40000

Joan Allen

32 3 48000

Bill Murray

45 3 50000

The DATA LIST command syntax to read the file is:

*skip_first_fixed.sps.
DATA LIST FIXED

FILE = '/examples/data/skip_first_fixed.txt'
RECORDS=2
SKIP=1

/name 1-20 (A)
/age 1-2 dept 4 salary 6-10.
EXECUTE.
The RECORDS subcommand indicates that there are two lines per case.
The skIP subcommand indicates that the first line of the file should not be included.

The first forward slash indicates the start of the list of variables contained on the first record
for each case. The only variable on the first record is the string variable name.

m  The second forward slash indicates the start of the variables contained on the second record
for each case.

Figure 3-10
Fixed-width, multiple-record text data file displayed in Data Editor
&) *Untitleds [] - Data Editor =2
File Edit Wiew Data Transform Analyze Graphs  Utlities  Add-ons  Window Help
12 : name
name | age | dept | salary | var var
1 {John Semith 26 2 40000
2|Joan Allen 32 3 43000
3| Bill Murray 45 3 50000
4
5 W
4/ v |\Data View £ variable View f |< >

Example

With fixed-width text data files, you can easily read selected portions of the data. For example,
using the skip_first fixed.txt data file from the above example, you could read just the age and
salary information.

*selected_vars_fixed.sps.
DATA LIST FIXED

FILE = '/examples/data/skip_first_fixed.txt'
RECORDS=2
SKIP=1
/2 age 1-2 salary 6-10.
EXECUTE.

B Asin the previous example, the command specifies that there are two records per case and that
the first line in the file should not be read.

m /2 indicates that variables should be read from the second record for each case. Since this is
the only list of variables defined, the information on the first record for each case is ignored,
and the employee’s name is not included in the data to be read.

m  The variables age and salary are read exactly as before, but no information is read from
columns 3-5 between those two variables because the command does not define a variable in
that space—so the department information is not included in the data to be read.

DATA LIST FIXED and Implied Decimals

If you specify a number of decimals for a numeric format with DATA LIST FIXED and some
data values for that variable do not contain decimal indicators, those values are assumed to
contain implied decimals.

Example

*implied_decimals.sps.

DATA LIST FIXED /varl (F5.2).
BEGIN DATA

123

123.0

1234

123.4

end data.

®  The values of 123 and 1234 will be read as containing two implied decimals positions,
resulting in values of 1.23 and 12.34.

®  The values of 123.0 and 123.4, however, contain explicit decimal indicators, resulting in
values of 123.0 and 123.4.

DATA LIST FREE (and LIST)and GET DATA /TYPE=TEXT do not read implied decimals; so a
value of 123 with a format of F5.2 will be read as 123.



40

Chapter 3

Text Data Files with Very Wide Records

Some machine-generated text data files with a large number of variables may have a single, very
wide record for each case. If the record width exceeds 8,192 columns/characters, you need to
specify the record length with the FILE HANDLE command before reading the data file.

*wide_file.sps.

*Read text data file with record length of 10,000.

*This command will stop at column 8,192.

DATA LIST FIXED
FILE='/examples/data/wide_file.txt"

/varl TO varl000 (1000F10).

EXECUTE.

*Define record length first.

FILE HANDLE wide_file NAME = '/examples/data/wide_file.txt'
/MODE = CHARACTER /LRECL = 10000.

DATA LIST FIXED
FILE = wide_file
/varl TO varl000

EXECUTE.

(1000F10) .

m  FEach record in the data file contains 1,000 10-digit values, for a total record length of 10,000
characters.

B The first DATA LIST command will read only the first 819 values (8,190 characters), and
the remaining variables will be set to the system-missing value. A warning message is
issued for each variable that is set to system-missing, which in this example means 181
warning messages.

B FILE HANDLE assigns a “handle” of wide file to the data file wide_file.txt.
B The LRECL subcommand specifies that each record is 10,000 characters wide.

B The FILE subcommand on the second DATA LIST command refers to the file handle
wide_file instead of the actual filename, and all 1,000 variables are read correctly.

Reading Different Types of Text Data

You can read text data recorded in a wide variety of formats. Some of the more common formats
are listed in the following table:

Type Example Format
specification

Numeric 123 F3

123.45 F6.2
Period as decimal indicator, comma as | 12,345 COMMAG6
thousands separator 12345 COMMA71
Comma as decimal indicator, period as | 123,4 DOT6
thousands separator 1.234,5 DOT7.1
Dollar $12,345 DOLLAR?7

$12,234.50 DOLLAR9.2
String (alphanumeric) Female A6
International date 28-OCT-1986 DATEI11
American date 10/28/1986 ADATEI10
Date and time 28 October, 1986 23:56 DATETIME22

For more information on date and time formats, see “Date and Time” in the “Universals” section
of the Command Syntax Reference. For a complete list of supported data formats, see “Variables”
in the “Universals” section of the Command Syntax Reference.

Example

*delimited_formats.sps.
DATA LIST LIST (" ")
/numericVar (F4) dotVar(DOT7.1) stringVar (a4) dateVar (DATE11l).
BEGIN DATA
1 2 abc 28/10/03
111 2.222,2 abcd 28-0CT-2003
111.11 222.222,222 abcdefg 28-October-2003

END DATA.
Figure 3-11
Different data types displayed in Data Editor
B *Untitled7 [] -Data Editor =Joed
Fle Edit Wiew Data Transform  Analvze  Graphs  Utliies Add-ons  Window Help
10 numericar
numericvar | dotvar | stringvar | date’ar | ~
1 1 20| abc 28-0CT-2003
2 111 2.2 2|abed 28-0CT-2003
3 11| 222222 2 abcd 28-0CT-2003
b
4 v \DataView f variahle View f |« > |

m  All of the numeric and date values are read correctly even if the actual values exceed the
maximum width (number of digits and characters) defined for the variables.

m  Although the third case appears to have a truncated value for numericVar, the entire value of
111.11 is stored internally. Since the defined format is also used as the display format, and
(F4) defines a format with no decimals, 111 is displayed instead of the full value. Values are
not actually truncated for display; they are rounded. A value of 111.99 would display as 112.

B The dateVar value of 28-October-2003 is displayed as 28-OCT-2003 to fit the defined width of
11 digits/characters.

m  For string variables, the defined width is more critical than with numeric variables. Any
string value that exceeds the defined width is truncated, so only the first four characters
for stringVar in the third case are read. Warning messages are displayed in the log for any
strings that exceed the defined width.

Reading Complex Text Data Files

“Complex” text data files come in a variety of flavors, including:

m  Mixed files in which the order of variables isn’t necessarily the same for all records and/or
some record types should be skipped entirely.

B Grouped files in which there are multiple records for each case that need to be grouped
together.

m  Nested files in which record types are related to each other hierarchically.

Mixed Files

A mixed file is one in which the order of variables may differ for some records and/or some
records may contain entirely different variables or information that shouldn’t be read.

Example

In this example, there are two record types that should be read: one in which state appears
before city and one in which city appears before state. There is also an additional record type
that shouldn’t be read.

*mixed_file.sps.
FILE TYPE MIXED RECORD = 1-2.
- RECORD TYPE 1.
- DATA LIST FIXED
/state 4-5 (A) city 7-17 (A) population 19-26 (F).
- RECORD TYPE 2.
- DATA LIST FIXED
/city 4-14 (A) state 16-17 (A) population 19-26 (F).
END FILE TYPE.
BEGIN DATA

01 TX Dallas 3280310
01 IL Chicago 8008507
02 Ancorage AK 257808
99 What am I doing here?
02 Casper WYy 63157
01 WI Madison 428563
END DATA.

B The commands that define how to read the data are all contained within the FILE TYPE-END
FILE TYPE structure.

MIXED identifies the type of data file.

RECORD = 1-2 indicates that the record type identifier appears in the first two columns
of each record.

B FEach DATA LIST command reads only records with the identifier value specified on the
preceding RECORD TYPE command. So if the value in the first two columns of the record is 1
(or 01), state comes before city, and if the value is 2, city comes before state.

B The record with the value 99 in the first two columns is not read, since there are no
corresponding RECORD TYPE and DATA LIST commands.

You can also include a variable that contains the record identifier value by including a variable
name on the RECORD subcommand of the FILE TYPE command, as in:

FILE TYPE MIXED /RECORD = recID 1-2.

You can also specify the format for the identifier value, using the same type of format
specifications as the DATA LIST command. For example, if the value is a string instead of a
simple numeric value:

FILE TYPE MIXED /RECORD = recID 1-2 (A).

Grouped Files

In a grouped file, there are multiple records for each case that should be grouped together based
on a unique case identifier. Each case usually has one record of each type. All records for a
single case must be together in the file.

Example

In this example, there are three records for each case. Each record contains a value that identifies
the case, a value that identifies the record type, and a grade or score for a different course.

* grouped_file.sps.

* A case is made up of all record types.
FILE TYPE GROUPED RECORD=6 CASE=student 1-4.
RECORD TYPE 1.

- DATA LIST /english 8-9 (A&).

RECORD TYPE 2.

- DATA LIST /reading 8-10.

RECORD TYPE 3.

- DATA LIST /math 8-10.

END FILE TYPE.

BEGIN DATA
0001 1 B+
0001 2 74
0001 3 83
0002 1 A
0002 3 71
0002 2 100
0003 1 B-
0003 2 88
0003 3 81
0004 1 C
0004 2 94
0004 3 91
END DATA.

B The commands that define how to read the data are all contained within the FILE TYPE-END
FILE TYPE structure.

B GROUPED identifies the type of data file.
B RECORD=6 indicates that the record type identifier appears in column 6 of each record.

B CASE=student 1-4 indicates that the unique case identifier appears in the first four columns
and assigns that value to the variable student in the active dataset.

®  The three RECORD TYPE and subsequent DATA LIST commands determine how each record
is read, based on the value in column 6 of each record.

Figure 3-12
Grouped data displayed in Data Editor
%) “Untitled8 [] - Data Editor =) |
File Edit “Wiew Data Transform Analyze Graphs Utilities Add-ons  Window Help
9 : student
student | english | reading | math | war va [
1 1B+ 74 a3
2 214 100 71
3 3B a8 a1
4 41 94 9
5
B
7 w
4/ v |\Data View £ variable View f < »
Example

In order to read a grouped data file correctly, all records for the same case must be contiguous

in the source text data file. If they are not, you need to sort the data file before reading it as a
grouped data file. You can do this by reading the file as a simple text data file, sorting it and saving
it, and then reading it again as a grouped file.

*grouped_file2.sps.

* Data file is sorted by record type instead of by
identification number.

DATA LIST FIXED
/alldata 1-80 (A) caseid 1-4.

BEGIN DATA

0001 1 B+
0002 1 A
0003 1 B-
0004 1 C
0001 2 74
0002 2 100
0003 2 88
0004 2 94
0001 3 83
0002 3 71
0003 3 81
0004 3 91
END DATA.

SORT CASES BY caseid.

WRITE OUTFILE='/temp/tempdata.txt'
/alldata.

EXECUTE.

* read the sorted file.

FILE TYPE GROUPED FILE='/temp/tempdata.txt'
RECORD=6 CASE=student 1-4.

- RECORD TYPE 1.

- DATA LIST /english 8-9 (A).

- RECORD TYPE 2.

- DATA LIST /reading 8-10.

- RECORD TYPE 3.

- DATA LIST /math 8-10.

END FILE TYPE.

EXECUTE.

m  The first DATA LIST command reads all of the data on each record as a single string variable.

In addition to being part of the string variable spanning the entire record, the first four columns
are read as the variable caseid.

The data file is then sorted by caseid, and the string variable alldata, containing all of the data

on each record, is written to the text file tempdata.txt.

m  The sorted file, tempdata.txt, is then read as a grouped data file, just like the inline data in
the previous example.

Prior to release 13.0, the maximum width of a string variable was 255 bytes. So in earlier releases,
for a file with records wider than 255 bytes, you would need to modify the job slightly to read and
write multiple string variables. For example, if the record width is 1,200:

DATA LIST FIXED
/stringl to string6 1-1200 (A) caseid 1-4.

This would read the file as six 200-byte string variables.

IBM® SPSS® Statistics can now handle much longer strings in a single variable: 32,767
bytes. So this workaround is unnecessary for release 13.0 or later. (If the record length exceeds
8,192 bytes, you need to use the FILE HANDLE command to specify the record length. See the
Command Syntax Reference for more information.)

Nested (Hierarchical) Files

In a nested file, the record types are related to each other hierarchically. The record types are
grouped together by a case identification number that identifies the highest level—the first
record type—of the hierarchy. Usually, the last record type specified—the lowest level of the
hierarchy—defines a case. For example, in a file containing information on a company’s sales
representatives, the records could be grouped by sales region. Information from higher record
types can be spread to each case. For example, the sales region information can be spread to the
records for each sales representative in the region.

Example

In this example, sales data for each sales representative are nested within sales regions (cities),
and those regions are nested within years.

*nested_filel.sps.

FILE TYPE NESTED RECORD=1 (A) .

- RECORD TYPE 'Y'.

- DATA LIST / Year 3-6.

- RECORD TYPE 'R'.

- DATA LIST / Region 3-13 (A).

- RECORD TYPE 'P'.

- DATA LIST / SalesRep 3-13 (A) Sales 20-23.
END FILE TYPE.

BEGIN DATA

Y 2002

R Chicago

P Jones 900
P Gregory 400
R Baton Rouge

P Rodriguez 300
P Smith 333
P Grau 100

END DATA.
Figure 3-13
Nested data displayed in Data Editor
] “Untitled? [] - Data Editor ==
File Edit “ew Data Transform Analvee Graphs  Utilities Add-ons  Window Help
8 Year
Year | Region | SalesRep | Sales | -
1 2002 |Chicago Jones 300
2 2002 |Chicago Gregary 400
3 2002 |Baton Rouge Rodriguez 300
4 2002 |Baton Rouge Smith 333
=3 2002 |Baton Rouge Grau 100
b
|| v |\ Data View 4 variahle view f [< 3]

B The commands that define how to read the data are all contained within the FILE TYPE-END
FILE TYPE structure.

B NESTED identifies the type of data file.
m  The value that identifies each record type is a string value in column 1 of each record.

B The order of the RECORD TYPE and associated DATA LIST commands defines the nesting
hierarchy, with the highest level of the hierarchy specified first. So, 'v' (year) is the highest
level, followed by 'R (region), and finally 'P' (person).

m  Eight records are read, but one of those contains year information and two identify regions; so,
the active dataset contains five cases, all with a value of 2002 for Year, two in the Chicago
Region and three in Baton Rouge.

Using INPUT PROGRAM to Read Nested Files

The previous example imposes some strict requirements on the structure of the data. For example,
the value that identifies the record type must be in the same location on all records, and it must
also be the same type of data value (in this example, a one-character string).

Instead of using a FILE TYPE structure, we can read the same data with an INPUT PROGRAM,
which can provide more control and flexibility.

Example

This first input program reads the same data file as the FILE TYPE NESTED example and obtains
the same results in a different manner.

* nested_inputl.sps.
INPUT PROGRAM.

- DATA LIST FIXED END=#eof /#type 1 (A).
- DO IF #eof.

- END FILE.

- END TIF.

- DO IF #type='Y'.

- REREAD.

- DATA LIST /Year 3-6.
- LEAVE Year.

- ELSE IF #type='R'.

- REREAD.

- DATA LIST / Region 3-13 (A).
- LEAVE Region.
- ELSE IF #type='P'.
- REREAD.
- DATA LIST / SalesRep 3-13 (A) Sales 20-23.
- END CASE.
- END IF.
END INPUT PROGRAM.
BEGIN DATA
Y 2002
R Chicago
P Jones 900
P Gregory 400
R Baton Rouge
Rodriguez 300
Smith 333
Grau 100
ND DATA.

P

P

P

E

B The commands that define how to read the data are all contained within the INPUT PROGRAM
structure.

m  The first DATA LIST command reads the temporary variable #fype from the first column
of each record.

B END=#eof creates a temporary variable named #eof that has a value of 0 until the end of the
data file is reached, at which point the value is set to 1.

B DO IF #eof evaluates as true when the value of #eof is set to 1 at the end of the file, and an
END FILE command is issued, which tells the INPUT PROGRAM to stop reading data. In this
example, this isn’t really necessary, since we’re reading the entire file; however, it will be
used later when we want to define an end point prior to the end of the data file.

B The second DO IF-ELSE IF-END IF structure determines what to do for each value of type.

B REREAD reads the same record again, this time reading either Year, Region, or SalesRep
and Sales, depending on the value of #type.

B LEAVE retains the value(s) of the specified variable(s) when reading the next record. So the
value of Year from the first record is retained when reading Region from the next record, and
both of those values are retained when reading SalesRep and Sales from the subsequent
records in the hierarchy. Thus, the appropriate values of Year and Region are spread to all of
the cases at the lowest level of the hierarchy.

B END CASE marks the end of each case. So, after reading a record with a #type value of 'P',
the process starts again to create the next case.

Example

In this example, the data file reflects the nested structure by indenting each nested level; so the
values that identify record type do not appear in the same place on each record. Furthermore,
at the lowest level of the hierarchy, the record type identifier is the last value instead of the
first. Here, an INPUT PROGRAM provides the ability to read a file that cannot be read correctly
by FILE TYPE NESTED.

*nested_input2.sps.
INPUT PROGRAM.
- DATA LIST FIXED END=#eof
/#yr 1 (A) #reg 3(A) #person 25 (A).

- DO IF #eof.

- END FILE.

- END IF.

- DO IF #yr='Y'.

- REREAD.

- DATA LIST /Year 3-6.

- LEAVE Year.

- ELSE IF #reg='R'.

- REREAD.

- DATA LIST / Region 5-15 (A).
- LEAVE Region.

- ELSE IF #person='P'.

- REREAD.

- DATA LIST / SalesRep 7-17 (A) Sales 20-23.
- END CASE.

- END TF.

END INPUT PROGRAM.

BEGIN DATA

Y 2002
R Chicago
Jones 900 P
Gregory 400 P
R Baton Rouge
Rodriguez 300 P
Smith 333 P
Grau 100 P
END DATA.

m  This time, the first DATA LIST command reads three temporary variables at different
locations, one for each record type.

B The DO IF-ELSE IF-END IF structure then determines how to read each record based on
the values of #yr, #reg, or #person.

B The remainder of the job is essentially the same as the previous example.

Example

Using the input program, we can also select a random sample of cases from each region and/or
stop reading cases at a specified maximum.

*nested_input3.sps.
INPUT PROGRAM.
NUMERIC #count (F8).
- DATA LIST FIXED END=#eof
/#yr 1 (A) #reg 3(A) #person 25 (A).
- DO IF #eof OR #count = 1000.
- END FILE.
- END IF.
- DO IF #yr='Y'.
- REREAD.
- DATA LIST /Year 3-6.
- LEAVE Year.
- ELSE IF #reg='R'.
- REREAD.
- DATA LIST / Region 5-15 (A).
- LEAVE Region.
- ELSE IF #person='P' AND UNIFORM(1000) < 500.
- REREAD.
- DATA LIST / SalesRep 7-17 (A) Sales 20-23.
- END CASE.
- COMPUTE #count=#count+1.
- END IF.
END INPUT PROGRAM.

BEGIN DATA

Y 2002
R Chicago
Jones 900 P
Gregory 400 P
R Baton Rouge
Rodriguez 300 P
Smith 333 P
Grau 100 P
END DATA.

B NUMERIC #count (F8) uses a scratch (temporary) variable as a case-counter variable.
Scratch variables are initialized to 0 and retain their values for subsequent cases.

B ELSE IF #person='P' AND UNIFORM(1000) < 500 will read a random sample of
approximately 50% from each region, since UNIFORM (1000) will generate a value less
than 500 approximately 50% of the time.

B COMPUTE #count=#count+1 increments the case counter by 1 for each case that is included.

B DO IF #eof OR #count = 1000 will issue an END FILE command if the case counter
reaches 1,000, limiting the total number of cases in the active dataset to no more than 1,000.

Since the source file must be sorted by year and region, limiting the total number of cases to 1,000
(or any value) may omit some years or regions within the last year entirely.

Repeating Data

In a repeating data file structure, multiple cases are constructed from a single record. Information
common to each case on the record may be entered once and then spread to all of the cases
constructed from the record. In this respect, a file with a repeating data structure is like a
hierarchical file, with two levels of information recorded on a single record rather than on separate
record types.

Example

In this example, we read essentially the same information as in the examples of nested file
structures, except now all of the information for each region is stored on a single record.

*repeating_data.sps.
INPUT PROGRAM.
DATA LIST FIXED
/Year 1-4 Region 6-16 (A) #numrep 19.
REPEATING DATA STARTS=22 /OCCURS=#numrep
/DATA=SalesRep 1-10 (A) Sales 12-14.
END INPUT PROGRAM.
BEGIN DATA

2002 Chicago 2 Jones 900Gregory 400
2002 Baton Rouge 3 Rodriguez 300Smith 333Grau 100
END DATA.

B The commands that define how to read the data are all contained within the INPUT PROGRAM
structure.

B The DATA LIST command defines two variables, Year and Region, that will be spread across
all of the cases read from each record. It also defines a temporary variable, #numrep.

B Onthe REPEATING DATA command, STARTS=22 indicates that the case starts in column 22.

B OCCURS=#numrep uses the value of the temporary variable, #numrep (defined on the previous
DATA LIST command), to determine how many cases to read from each record. So, two
cases will be read from the first record, and three will be read from the second.

®  The DATA subcommand defines two variables for each case. The column locations for those
variables are relative locations. For the first case, column 22 (specified on the STARTS
subcommand) is read as column 1. For the next case, column 1 is the first column after
the end of the defined column span for the last variable in the previous case, which would
be column 36 (22+14=36).

The end result is an active dataset that looks remarkably similar to the data file created from the
hierarchical source data file.

Figure 3-14
Repeating data displayed in Data Editor
# *Untitled10 [] - Data Editor o
File Edit “iew Data Transform Analvze Graphs  Ukilities Add-ons  Window Help
11 Year
Year | Region | SalesRep | Sales | var s
1| 2002 |Chicaga Jones 500 )
2 2002 |Chicago Gregary 400
3 2002 |Baton Rouge Rodriguez 300
4 2002 |Baton Rouge Srnith 333
5 2002 |Baton Rouge Grau 100
E |
7 b
« v \DataView £ variable view f <1 [>]]

Reading SAS Data Files

IBM® SPSS® Statistics can read the following types of SAS files:
m  SAS long filename, versions 7 through 9

SAS short filenames, versions 7 through 9

SAS version 6 for Windows

SAS version 6 for UNIX

SAS Transport

The basic structure of a SAS data file is very similar to a data file in SPSS Statistics format—rows
are cases (observations), and columns are variables—and reading SAS data files requires only a
single, simple command: GET SAS.

Example

In its simplest form, the GET SAS command has a single subcommand that specifies the SAS
filename.

*get_sas.sps.
GET SAS DATA='/examples/data/demo.sd7'.

m  SAS variable names that do not conform to SPSS Statistics variable-naming rules are
converted to valid variable names.

m  SAS variable labels specified on the LABEL statement in the DATA step are used as variable
labels in SPSS Statistics.

Example

SAS value formats are similar to SPSS Statistics value labels, but SAS value formats are saved in
a separate file; so if you want to use value formats as value labels, you need to use the FORMATS
subcommand to specify the formats file.

GET SAS DATA='datafilename.sd7'
/FORMATS="'formatsfilename.sd7"'.

m  Labels assigned to single values are retained.
m Labels assigned to a range of values are ignored.

m Labels assigned to the SAS keywords L.OwW, HIGH, and OTHER are ignored.

The file specified on the FORMATS subcommand must be a SAS-format catalog file created with
the proc format command. For example:

libname mylib 'c:\mydir\' ;

proc format library = mylib ;
value YesNo

='No"

='Yes' ;
value HighLow

1 'Low'

2 'Medium'

3 '"High' ;

options fmtsearch=(mylib) ;

proc datasets library = mylib ;
modify mydata;

format varl var2 var3 YesNo.;
format vard var5 var6 HighLow.;
quit;

B  1ibname defines a “library,” which is a directory path.

B proc format defines two formats: YesNo and HighLow. Each format defines a set of value
labels associated with data values.

B proc datasets identifies the data file—mydata—and the variables to which each of the
defined formats should be applied. So the YesNo format is applied to variables varl, var2, and
var3, and the HighLow format is applied to the variables var4, var5, and var6.

B This creates the SAS catalog file c:\mydir\formats.sas7bcat.

Reading Stata Data Files

GET STATA reads Stata-format data files created by Stata versions 4 through 8. The only
specification is the FILE keyword, which specifies the Stata data file to be read.

m Variable names. Stata variable names are converted to IBM® SPSS® Statistics variable names
in case-sensitive form. Stata variable names that are identical except for case are converted

to valid variable names by appending an underscore and a sequential letter ( 4, B, C, ...,
_Z, AA, AB, ..., and so forth).

m Variable labels. Stata variable labels are converted to SPSS Statistics variable labels.

m  Value labels. Stata value labels are converted to SPSS Statistics value labels, except for Stata
value labels assigned to “extended” missing values.

m  Missing values. Stata “extended” missing values are converted to system-missing values.

m  Date conversion. Stata date format values are converted to SPSS Statistics DATE format
(d-m-y) values. Stata “time-series” date format values (weeks, months, quarters, and so on)
are converted to simple numeric (F) format, preserving the original, internal integer value,
which is the number of weeks, months, quarters, and so on, since the start of 1960.

Example

GET STATA FILE='/examples/data/statafile.dta'.

Code Page and Unicode Data Sources

Starting with release 16.0, you can read and write Unicode data files.

SET UNICODE NO|YES controls the default behavior for determining the encoding for reading
and writing data files and syntax files.

NO. Use the current locale setting to determine the encoding for reading and writing data and
command syntax files. This is referred to as code page mode. This is the default. The alias is OFF.

YES. Use Unicode encoding (UTF-8) for reading and writing data and command syntax files. This
is referred to as Unicode mode. The alias is ON.

B You can change the UNICODE setting only when there are no open data sources.

B The UNICODE setting persists across sessions and remains in effect until it is explicitly
changed.

There are a number of important implications regarding Unicode mode and Unicode files:

m Data and syntax files saved in Unicode encoding should not be used in releases prior to 16.0.
For syntax files, you can specify local encoding when you save the file. For data files, you
should open the data file in code page mode and then resave it if you want to read the file
with earlier versions.

B When code page data files are read in Unicode mode, the defined width of all string variables
is tripled.

B  The GET command determines the file encoding for IBM® SPSS® Statistics data files from
the file itself, regardless of the current mode setting (and defined string variable widths in
code page files are tripled in Unicode mode).

®m  For text data files read with DATA LIST and related commands (for example, REPEATING
DATA and FILE TYPE) or written with PRINT or WRITE, you can override the default
encoding with the ENCODING subcommand.

B GET DATA uses the default encoding for reading text data files (TYPE=TXT), which is UTF-8
in Unicode mode or the code page determined by the current locale in code page mode.

m  OMS uses default encoding for writing text files (FORMAT=TEXT and FORMAT=TABTEXT) and
for writing SPSS Statistics data files (FORMAT=SAV).

GET TRANSLATE reads data in the current locale code page, regardless of mode.

By default, GET STATA, GET SAS, and SAVE TRANSLATE read and write data in the current
locale code page, regardless of mode. You can use the ENCODING subcommand to specify a
different encoding.

B By default, SAVE TRANSLATE saves SAS 9 files in UTF-8 format in Unicode mode and in the
current locale code page in code page mode. You can use the ENCODING subcommand to
specify a different encoding.

®  For syntax files run via INCLUDE or INSERT, you can override the default encoding with the
ENCODING subcommand.

m  For syntax files, the encoding is set to Unicode after execution of the block of commands that
includes SET UNICODE=YES. You must run SET UNICODE=YES separately from subsequent
commands that contain Unicode characters not recognized by the local encoding in effect
prior to switching to Unicode.

Example: Reading Code Page Text Data in Unicode Mode

*read_codepage.sps.

CD '/examples/data'.

DATASET CLOSE ALL.

NEW FILE.

SET UNICODE YES.

DATA LIST LIST FILE='codepage.txt'
/NumVar (F3) StringVar (AS8).

EXECUTE.

DATA LIST LIST FILE='codepage.txt' ENCODING='Locale'
/NumVar (F3) StringVar (A8).

COMPUTE ByteLength=LENGTH (RTRIM(StringVar)) .

COMPUTE CharLength=CHAR.LENGTH (StringVar) .

SUMMARIZE
/TABLES=StringVar ByteLength CharLength
/FORMAT=VALIDLIST /CELLS=COUNT
/TITLE='Unicode Byte and Character Counts'.

DISPLAY DICTIONARY VARIABLES=StringVar.

DATASET CLOSE ALL.

NEW FILE.

SET UNICODE NO.

B SET UNICODE YES switches from the default code page mode to Unicode mode. Since you
can change modes only when there are no open data sources, this is preceded by DATASET
CLOSE ALL to close all named datasets and NEW FILE to replace the active dataset with a
new, empty dataset.

m  The text data file codepage.txt is a code page file, not a Unicode file; so any string values that
contain anything other than 7-bit ASCII characters will be read incorrectly when attempting
to read the file as if it were Unicode. In this example, the string value résumé contains two
accented characters that are not 7-bit ASCII.

B The first DATA LIST command attempts to read the text data file in the default encoding. In

Unicode mode, the default encoding is Unicode (UTF-8), and the string value résumé cannot
be read correctly, which generates a warning:

>Warning # 1158

>An invalid character was encountered in a field read under an A format. In
>double-byte data such as Japanese, Chinese, or Korean text, this could be
>caused by a single character being split between two fields. The character
>will be treated as an exclamation point.

ENCODING='Locale' on the second DATA LIST command identifies the encoding for the
text data file as the code page for the current locale, and the string value résumé is read
correctly. (If your current locale is not English, use ENCODING="'1252".)

LENGTH (RTRIM (StringVar) ) returns the number of bytes in each value of StringVar. Note
that résumeé is eight bytes in Unicode mode because each accented e takes two bytes.

CHAR.LENGTH (StringVar) returns the number characters in each value of Stringlar.
While an accented e is two bytes in Unicode mode, it is only one character; so both résumé
and resume contain six characters.

The output from the DISPLAY DICTIONARY command shows that the defined width of
StringVar has been tripled from the input width of A8 to A24. To minimize the expansion of
string widths when reading code page data in Unicode mode, you can use the ALTER TYPE
command to automatically set the width of each string variable to the maximum observed
string value for that variable. For more information, see the topic “Changing Data Types
and String Widths” in Chapter 6 on p. 95.

Figure 3-15
String width in Unicode mode

File Operations

You can combine and manipulate data sources in a number of ways, including:
Using multiple data sources

Merging data files

Aggregating data

Weighting data

Changing file structure

Using output as input (For more information, see “Using Output as Input with OMS” in
Chapter 9 on p. 143.)

Using Multiple Data Sources

Starting with release 14.0, you can have multiple data sources open at the same time.

®  When you use the dialog boxes and wizards in the graphical user interface to open data
sources, the default behavior is to open each data source in a new Data Editor window, and
any previously open data sources remain open and available for further use. You can change
the active dataset simply by clicking anywhere in the Data Editor window of the data source
that you want to use or by selecting the Data Editor window for that data source from the
Window menu.

B In command syntax, the default behavior remains the same as in previous releases: reading a
new data source automatically replaces the active dataset. If you want to work with multiple
datasets using command syntax, you need to use the DATASET commands.

The DATASET commands (DATASET NAME, DATASET ACTIVATE, DATASET DECLARE,
DATASET COPY, DATASET CLOSE) provide the ability to have multiple data sources open at the

same time and control which open data source is active at any point in the session. Using defined
dataset names, you can then:

m  Merge data (for example, MATCH FILES, ADD FILES, UPDATE) from multiple different
source types (for example, text data, database, spreadsheet) without saving each one as an
external IBM® SPSS® Statistics data file first.

m  Create new datasets that are subsets of open data sources (for example, males in one subset,
females in another, people under a certain age in another, or original data in one set and
transformed/computed values in another subset).

m  Copy and paste variables, cases, and/or variable properties between two or more open data
sources in the Data Editor.

© Copyright SPSS Inc. 1989, 2010 55

m  Commands operate on the active dataset. The active dataset is the data source most recently
opened (for example, by commands such as GET DATA, GET SAS, GET STATA, GET
TRANSLATE) or most recently activated by a DATASET ACTIVATE command.

Note: The active dataset can also be changed by clicking anywhere in the Data Editor window
of an open data source or selecting a dataset from the list of available datasets in a syntax
window toolbar.

B Variables from one dataset are not available when another dataset is the active dataset.

m  Transformations to the active dataset—before or after defining a dataset name—are preserved
with the named dataset during the session, and any pending transformations to the active
dataset are automatically executed whenever a different data source becomes the active
dataset.

® Dataset names can be used in most commands that can contain references to SPSS Statistics
data files.

m  Wherever a dataset name, file handle (defined by the FILE HANDLE command), or filename
can be used to refer to SPSS Statistics data files, defined dataset names take precedence
over file handles, which take precedence over filenames. For example, if file/ exists as both
a dataset name and a file handle, FILE=filel in the MATCH FILES command will be
interpreted as referring to the dataset named filel, not the file handle.

Example

*multiple_datasets.sps.

DATA LIST FREE /filelVar.

BEGIN DATA

11 12 13

END DATA.

DATASET NAME filel.

COMPUTE filelVar=MOD(filelVar,10).

DATA LIST FREE /file2Var.

BEGIN DATA

21 22 23

END DATA.

DATASET NAME file2.

*file2 is now the active dataset; so the following
command will generate an error.

FREQUENCIES VARIABLES=filelVar.

*now activate dataset filel and rerun Frequencies.

DATASET ACTIVATE filel.

FREQUENCIES VARIABLES=filelVar.

m  The first DATASET NAME command assigns a name to the active dataset (the data defined
by the first DATA 1LIST command). This keeps the dataset open for subsequent use in the
session after other data sources have been opened. Without this command, the dataset would
automatically close when the next command that reads/opens a data source is run.

®  The cOMPUTE command applies a transformation to a variable in the active dataset. This
transformation will be preserved with the dataset named file/. The order of the DATASET
NAME and COMPUTE commands is not important. Any transformations to the active dataset,
before or after assigning a dataset name, are preserved with that dataset during the session.

B The second DATA LIST command creates a new dataset, which automatically becomes the
active dataset. The subsequent FREQUENCIES command that specifies a variable in the first
dataset will generate an error, because file/ is no longer the active dataset, and there is no
variable named file! Var in the active dataset.

B DATASET ACTIVATE makes filel the active dataset again, and now the FREQUENCIES
command will work.

Example

*dataset_subsets.sps.
DATASET CLOSE ALL.

DATA LIST FREE /gender.
BEGIN DATA
0011011100

END DATA.

DATASET NAME original.
DATASET COPY males.
DATASET ACTIVATE males.
SELECT IF gender=0.
DATASET ACTIVATE original.
DATASET COPY females.
DATASET ACTIVATE females.
SELECT IF gender=1.
EXECUTE.

B The first DATASET COPY command creates a new dataset, males, that represents the state of
the active dataset at the time it was copied.

B The males dataset is activated and a subset of males is created.
m  The original dataset is activated, restoring the cases deleted from the males subset.

m  The second DATASET COPY command creates a second copy of the original dataset with the
name females, which is then activated and a subset of females is created.

m  Three different versions of the initial data file are now available in the session: the original
version, a version containing only data for males, and a version containing only data for
females.

Figure 4-1
Multiple subsets available in the same session

Merging Data Files

You can merge two or more datasets in several ways:
m  Merge datasets with the same cases but different variables
®  Merge datasets with the same variables but different cases

m  Update values in a master data file with values from a transaction file

Merging Files with the Same Cases but Different Variables

The MATCH FILES command merges two or more data files that contain the same cases but
different variables. For example, demographic data for survey respondents might be contained
in one data file, and survey responses for surveys taken at different times might be contained in
multiple additional data files. The cases are the same (respondents), but the variables are different
(demographic information and survey responses).

This type of data file merge is similar to joining multiple database tables except that you are
merging multiple IBM® SPSS® Statistics data files rather than database tables. For information on
reading multiple database tables with joins, see “Reading Multiple Tables” in Chapter 3 on p. 22.

One-to-One Matches

The simplest type of match assumes that there is basically a one-to-one relationship between cases
in the files being merged—for each case in one file, there is a corresponding case in the other file.

Example

This example merges a data file containing demographic data with another file containing survey
responses for the same cases.

*match_filesl.sps.
*first make sure files are sorted correctly.
GET FILE='/examples/data/match_responsel.sav'.
SORT CASES BY id.
DATASET NAME responses.
GET FILE='/examples/data/match_demographics.sav'.
SORT CASES BY id.
*now merge the survey responses with the demographic info.
MATCH FILES /FILE=*
/FILE=responses
/BY id.
EXECUTE.

B DATASET NAME is used to name the first dataset, so it will remain available after the second
dataset is opened.

B SORT CASES BY idis used to sort both datasets in the same case order. Cases are merged
sequentially, so both datasets must be sorted in the same order to make sure that cases are
merged correctly.

B MATCH FILES merges the two datasets. FILE=* indicates the active dataset (the demographic
dataset).

®  The BY subcommand matches cases by the value of the ID variable in both datasets. In this
example, this is not technically necessary, since there is a one-to-one correspondence between
cases in the two datasets and the datasets are sorted in the same case order. However, if
the datasets are not sorted in the same order and no key variable is specified on the BY
subcommand, the datasets will be merged incorrectly with no warnings or error messages;
whereas, if a key variable is specified on the BY subcommand and the datasets are not sorted
in the same order of the key variable, the merge will fail and an appropriate error message
will be displayed. If the datasets contain a common case identifier variable, it is a good
practice to use the BY subcommand.

B Any variables with the same name are assumed to contain the same information, and only
the variable from the first dataset specified on the MATCH FILES command is included in
the merged dataset. In this example, the ID variable (id) is present in both datasets, and the
merged dataset contains the values of the variable from the demographic dataset—which
is the first dataset specified on the MATCH FILES command. (In this case, the values are
identical anyway.)

m  For string variables, variables with the same name must have the same defined width in both
files. If they have different defined widths, an error results and the command does not run.
This includes string variables used as BY variables.

Example

Expanding the previous example, we will merge the same two data files plus a third data file
that contains survey responses from a later date. Three aspects of this third file warrant special
attention:

m  The variable names for the survey questions are the same as the variable names in the survey
response data file from the earlier date.

B One of the cases that is present in both the demographic data file and the first survey response
file is missing from the new survey response data file.

m  The source file is not in IBM® SPSS® Statistics format; it’s an Excel worksheet.

*match_files2.sps.
GET FILE='/examples/data/match_responsel.sav'.
SORT CASES BY id.
DATASET NAME responsel.
GET DATA /TYPE=XLS
/FILE="'/examples/data/match_response2.xls'.
SORT CASES BY id.
DATASET NAME response2.
GET FILE='/examples/data/match_demographics.sav'.
SORT CASES BY id.
MATCH FILES /FILE=*
/FILE=responsel
/FILE=response?2
/RENAME opinionl=opinionl_2 opinion2=opinion2_2
opinion3=opinion3_2 opinion4=opinion4_2
/BY id.
EXECUTE.

As before, all of the datasets are sorted by the values of the ID variable.

MATCH FILES specifies three datasets this time: the active dataset that contains the
demographic information and the two datasets containing survey responses from two different
dates.

® The RENAME command after the FILE subcommand for the second survey response dataset
provides new names for the survey response variables in that dataset. This is necessary to
include these variables in the merged dataset. Otherwise, they would be excluded because the
original variable names are the same as the variable names in the first survey response dataset.

®  The BY subcommand is necessary in this example because one case (id = 184) is missing
from the second survey response dataset, and without using the BY variable to match cases,
the datasets would be merged incorrectly.

B All cases are included in the merged dataset. The case missing from the second survey
response dataset is assigned the system-missing value for the variables from that dataset
(opinionl 2—opiniond_2).

Figure 4-2

Merged files displayed in Data Editor

. ] *match_demographics.sav [] - Data Editor E]@
File Edit WYiew Data Transform Analyze Graphs Utilities Add-ons  Window Help

13:id Yisible: 13 of 13 Var
it ‘Age Gender

Income_
category|
1[1580| 55 |m 3
21170 29/f
3|184 | 42f
41216 39|F
51227 | B2|m
B
7
8

Religion| opiniont| opinion2| opinion3| opiniond| opinion?_2

0pini0n2_2‘ opinion3_2 ‘ opiniond_2 ‘A

2 3 2
2 2

228 24/f
272| 25|f
295|500 |f
91333 30|m
101385 23|m
111391 58 |m

< » \\Data View £ variable View / 3

= | RO 00| D | D | DD e

0| e | D e | D RO | D | RO
LA R SRR Gy i R R U R R U i TR R SRS |
S Un R TR R & R SN R U R R S
M 0| k| QD e | = (| RO QD RO DD
QR QD | QD M| Q)| = = (| —
| | | QO RO | QD] 0D

PO = Q0 | | QO 0D

| 0| QI QI | b | |

¥Moelwlw o e e =]




61

File Operations
Table Lookup (One-to-Many) Matches

A table lookup file is a file in which data for each case can be applied to multiple cases in the
other data file(s). For example, if one file contains information on individual family members
(such as gender, age, education) and the other file contains overall family information (such as
total income, family size, location), you can use the file of family data as a table lookup file and
apply the common family data to each individual family member in the merged data file.

Specifying a file with the TABLE subcommand instead of the FTLE subcommand indicates that
the file is a table lookup file. The following example merges two text files, but they could be any
combination of data source formats. For information on reading different types of data, see
Chapter 3 on p. 19.

*match_table_lookup.sps.

DATA LIST LIST
FILE='/examples/data/family_ data.txt'

/household_id total_income family_size region.

SORT CASES BY household_id.

DATASET NAME household.

DATA LIST LIST
FILE='/examples/data/individual_data.txt'
/household_id indv_id age gender education.

SORT CASE BY household_id.

DATASET NAME individual.

MATCH FILES TABLE='household'
/FILE="'individual'

/BY household_id.

EXECUTE.

Merging Files with the Same Variables but Different Cases

The ADD FILES command merges two or more data files that contain the same variables but
different cases. For example, regional revenue for two different company divisions might be
stored in two separate data files. Both files have the same variables (region indicator and revenue)
but different cases (each region for each division is a case).

Example

ADD FILES relies on variable names to determine which variables represent the “same” variables
in the data files being merged. In the simplest example, all of the files contain the same set of
variables, using the exact same variable names, and all you need to do is specify the files to be
merged. In this example, the two files both contain the same two variables, with the same two
variable names: Region and Revenue.

*add_filesl.sps.

ADD FILES
/FILE = '/examples/data/catalog.sav'
/FILE ='/examples/data/retail.sav'
/IN = Division.
EXECUTE.

VALUE LABELS Division 0 'Catalog' 1 'Retail Store'.

Figure 4-3
Cases from one file added to another file

B Cases are added to the active dataset in the order in which the source data files are specified on
the ADD FILES command; all of the cases from catalog.sav appear first, followed by all of
the cases from retail.sav.

B The 1IN subcommand after the FILE subcommand for retail.sav creates a new variable
named Division in the merged dataset, with a value of 1 for cases that come from retail.sav
and a value of 0 for cases that come from catalog.sav. (If the IN subcommand was placed
immediately after the FILE subcommand for catalog.sav, the values would be reversed.)

B The VALUE LABELS command provides descriptive labels for the Division values of 0 and 1,
identifying the division for each case in the merged dataset.

Example

Now that we’ve had a good laugh over the likelihood that all of the files have the exact same
structure with the exact same variable names, let’s look at a more realistic example. What if the
revenue variable had a different name in one of the files and one of the files contained additional
variables not present in the other files being merged?

*add_files2.sps.

***first throw some curves into the data***.
GET FILE = '/examples/data/catalog.sav'.
RENAME VARIABLES (Revenue=Sales).

DATASET NAME catalog.

GET FILE = '/examples/data/retail.sav'.
COMPUTE ExtraVar = 9.
EXECUTE.

DATASET NAME retail.
***show default behavior***.

ADD FILES
/FILE = 'catalog'
/FILE = 'retail'
/IN = Division.

EXECUTE.

***now treat Sales and Revenue as same variable***,
***and drop ExtraVar from the merged file***,

ADD FILES
/FILE = 'catalog'
/RENAME (Sales = Revenue)
/FILE = 'retail'

/IN = Division

/DROP ExtraVar

/BY Region.

EXECUTE.

All of the commands prior to the first ADD FILES command simply modify the original data
files to contain minor variations—Revenue is changed to Sales in one data file, and an extra
variable, ExtraVar, is added to the other data file.

The first ADD FILES command is similar to the one in the previous example and shows the
default behavior if nonmatching variable names and extraneous variables are not accounted
for—the merged dataset has five variables instead of three, and it also has a lot of missing
data. Sales and Revenue are treated as different variables, resulting in half of the cases having
values for Sales and half of the cases having values for Revenue—and cases from the second
data file have values for ExtraVar, but cases from the first data file do not, since this variable
does not exist in that file.

Figure 4-4
Probably not what you want when you add cases from another file

In the second ADD FILES command, the RENAME subcommand after the FILE subcommand
for catalog will treat the variable Sales as if its name were Revenue, so the variable name will
match the corresponding variable in retail.

The DROP subcommand following the FILE subcommand for femp2.sav (and the associated
1IN subcommand) will exclude ExtraVar from the merged dataset. (The DROP subcommand
must come after the FILE subcommand for the file that contains the variables to be excluded.)

The BY subcommand adds cases to the merged data file in ascending order of values of the
variable Region instead of adding cases in file order—but this requires that both files already
be sorted in the same order of the BY variable.

Figure 4-5
Cases added in order of Region variable instead of file order

Updating Data Files by Merging New Values from Transaction Files

You can use the UPDATE command to replace values in a master file with updated values recorded
in one or more files called transaction files.

*update.sps.

GET FILE = '/examples/data/update_transaction.sav'.
SORT CASE BY id.

DATASET NAME transaction.

GET FILE = '/examples/data/update_master.sav'.

SORT CASES BY id.

UPDATE /FILE = *

/FILE = transaction
/IN = updated
/BY id.
EXECUTE.

B SORT CASES BY idisused to sort both files in the same case order. Cases are updated
sequentially, so both files must be sorted in the same order.

B The first FILE subcommand on the UPDATE command specifies the master data file. In this
example, FILE = * specifies the active dataset.

®  The second FILE subcommand specifies the dataset name assigned to the transaction file.

®  The IN subcommand immediately following the second FILE subcommand creates a new
variable called updated in the master data file; this variable will have a value of 1 for any
cases with updated values and a value of 0 for cases that have not changed.

m  The BY subcommand matches cases by id. This subcommand is required. Transaction files
often contain only a subset of cases, and a key variable is necessary to match cases in the
two files.

Figure 4-6

Original file, transaction file, and update file

The salary values for the cases with the id values of 103 and 201 are both updated.

The department value for case 201 is updated, but the department value for case 103 is not
updated. System-missing values in the transaction files do not overwrite existing values in the
master file, so the transaction files can contain partial information for each case.

Aggregating Data

The AGGREGATE command creates a new dataset where each case represents one or more cases
from the original dataset. You can save the aggregated data to a new dataset or replace the active
dataset with aggregated data. You can also append the aggregated results as new variables to

the current active dataset.

Example

In this example, information was collected for every person living in a selected sample of
households. In addition to information for each individual, each case contains a variable that
identifies the household. You can change the unit of analysis from individuals to households by
aggregating the data based on the value of the household ID variable.

*aggregatel.sps.

***create some sample data***.

DATA LIST FREE (" ")
/ID_household (F3)

BEGIN DATA

ID_person

(F2)

Income

(F8) .

101 1 12345 101 2 47321 101 3 500 101 4 O
102 1 77233 102 2 0
103 1 19010 103 2 98277 103 3 0
104 1 101244
END DATA.
***now aggregate based on household id***.
AGGREGATE

/JOUTFILE = * MODE = REPLACE
/BREAK = ID_household
/Household_Income = SUM(Income)
/Household_Size = N.

OUTFILE = * MODE = REPLACE replaces the active dataset with the aggregated data.
BREAK = ID_household combines cases based on the value of the household ID variable.

Household_Income = SUM(Income) creates a new variable in the aggregated dataset
that is the total income for each household.

B Household_Size = N creates a new variable in the aggregated dataset that is the number
of original cases in each aggregated case.

Figure 4-7
Original and aggregated data

Example

You can also use MODE = ADDVARIABLES to add group summary information to the original
data file. For example, you could create two new variables in the original data file that contain
the number of people in the household and the per capita income for the household (total income
divided by number of people in the household).

*aggregate2.sps.
DATA LIST FREE (" ")
/ID_household (F3) ID_person (F2) Income (F8).
BEGIN DATA
101 1 12345 101 2 47321 101 3 500 101 4 ©
102 1 77233 102 2 0
103 1 19010 103 2 98277 103 3 0
104 1 101244
END DATA.
AGGREGATE

/OUTFILE = * MODE = ADDVARIABLES
/BREAK = ID_household
/per_capita_Income = MEAN (Income)
/Household_Size = N.

m  As with the previous example, OUTFILE = * specifies the active dataset as the target for the
aggregated results.
m Instead of replacing the original data with aggregated data, MODE = ADDVARIABLES will
add aggregated results as new variables to the active dataset.
®  As with the previous example, cases will be aggregated based on the household ID value.
®  The MEAN function will calculate the per capita household incomes.
Figure 4-8
Aggregate summary data added to original data

Aggregate Summary Functions

The new variables created when you aggregate a data file can be based on a wide variety of numeric
and statistical functions applied to each group of cases defined by the BREAK variables, including:

Number of cases in each group

Sum, mean, median, and standard deviation

Minimum, maximum, and range

Percentage of cases between, above, and/or below specified values
First and last nonmissing value in each group

Number of missing values in each group

For a complete list of aggregate functions, see the AGGREGATE command in the Command Syntax
Reference.

Weighting Data

The WEIGHT command simulates case replication by treating each case as if it were actually the
number of cases indicated by the value of the weight variable. You can use a weight variable to
adjust the distribution of cases to more accurately reflect the larger population or to simulate raw
data from aggregated data.

Example

A sample data file contains 52% males and 48% females, but you know that in the larger
population the real distribution is 49% males and 51% females. You can compute and apply a
weight variable to simulate this distribution.

*weight_sample.sps.

***create sample data of 52 males, 48 females***.
NEW FILE.

INPUT PROGRAM.

- STRING gender (A6).

- LOOP #I =1 TO 100.

- DO IF #I <= 52.

- COMPUTE gender='Male’'.

- ELSE.

- COMPUTE Gender='Female'.

- END TF.

- COMPUTE AgeCategory = trunc(uniform(3)+1).
- END CASE.

- END LOOP.

- END FILE.

END INPUT PROGRAM.

FREQUENCIES VARIABLES=gender AgeCategory.
***create and apply weightvar***,

***to simulate 49 males, 51 females***.
DO IF gender = 'Male’'.

- COMPUTE weightvar=49/52.

ELSE IF gender = 'Female'.

- COMPUTE weightvar=51/48.

END TIF.

WEIGHT BY weightvar.

FREQUENCIES VARIABLES=gender AgeCategory.

®  Everything prior to the first FREQUENCIES command simply generates a sample dataset with
52 males and 48 females.

m The DO IF structure sets one value of weightvar for males and a different value for females.
The formula used here is: desired proportion/observed proportion. For males, it is 49/52
(0.94), and for females, it is 51/48 (1.06).

B The WEIGHT command weights cases by the value of weightvar, and the second FREQUENCIES
command displays the weighted distribution.

Note: In this example, the weight values have been calculated in a manner that does not alter the
total number of cases. If the weighted number of cases exceeds the original number of cases, tests
of significance are inflated; if it is smaller, they are deflated. More flexible and reliable weighting
techniques are available in the Complex Samples add-on module.

Example

You want to calculate measures of association and/or significance tests for a crosstabulation, but
all you have to work with is the summary table, not the raw data used to construct the table. The
table looks like this:

Male Female Total
Under $50K 25 35 60
$50K+ 30 10 40
Total 55 45 100

You then read the data using rows, columns, and cell counts as variables; then, use the cell count
variable as a weight variable.

*weight.sps.
DATA LIST LIST /Income Gender count.
BEGIN DATA

1, 1, 25
1, 2, 35
2, 1, 30
2, 2, 10
END DATA.

VALUE LABELS
Income 1 'Under $50K' 2 'S$50K+'
/Gender 1 'Male' 2 'Female'.

WEIGHT BY count.

CROSSTABS TABLES=Income by Gender
/STATISTICS=CC PHI.

m  The values for Income and Gender represent the row and column positions from the original
table, and count is the value that appears in the corresponding cell in the table. For example,
1, 2, 35 indicates that the value in the first row, second column is 35. (The Total row and
column are not included.)

® The VALUE LABELS command assigns descriptive labels to the numeric codes for Income and
Gender. In this example, the value labels are the row and column labels from the original table.

B The WEIGHT command weights cases by the value of count, which is the number of cases
in each cell of the original table.

B The crROSSTABS command produces a table very similar to the original and provides statistical
tests of association and significance.

Figure 4-9

Crosstabulation and significance tests for reconstructed table

Income * Gender Crosstabulation

Gender

Total 55 45 100

Changing File Structure

IBM® SPSS® Statistics expects data to be organized in a certain way, and different types of
analysis may require different data structures. Since your original data can come from many
different sources, the data may require some reorganization before you can create the reports

or analyses that you want.

Transposing Cases and Variables

You can use the FLIP command to create a new data file in which the rows and columns in the
original data file are transposed so that cases (rows) become variables and variables (columns)

become cases.

Example

Although IBM® SPSS® Statistics expects cases in the rows and variables in the columns,
applications such as Excel don’t have that kind of data structure limitation. So what do you do with
an Excel file in which cases are recorded in the columns and variables are recorded in the rows?

Figure 4-10

Excel file with cases in columns, variables in rows

Here are the commands to read the Excel spreadsheet and transpose the rows and columns:

*flip_excel.sps.
GET DATA /TYPE=XLS
/FILE="'/examples/data/flip_excel.xls'

/READNAMES=0N
FLIP VARIABLES=Newton Boris Kendall Dakota Jasper Maggie

/NEWNAME=V1.
RENAME VARIABLES (CASE_LBL = Name) .

B READNAMES=ON in the GET DATA command reads the first row of the Excel spreadsheet as
variable names. Since the first cell in the first row is blank, it is assigned a default variable

name of V1.

® The FLIP command creates a new active dataset in which all of the variables specified will
become cases and all cases in the file will become variables.

m  The original variable names are automatically stored as values in a new variable called
CASE LBL. The subsequent RENAME VARIABLES command changes the name of this
variable to Name.

B NEWNAME=V1 uses the values of variable V'] as variable names in the transposed data file.

Figure 4-11
Original and transposed data in Data Editor

Cases to Variables

Sometimes you may need to restructure your data in a slightly more complex manner than simply

flipping rows and columns.
Many statistical techniques in IBM® SPSS® Statistics are based on the assumption that cases
(rows) represent independent observations and/or that related observations are recorded in separate

variables rather than separate cases. If a data file contains groups of related cases, you may not be

able to use the appropriate statistical techniques (for example, the paired samples 7 test or repeated
measures GLM) because the data are not organized in the required fashion for those techniques.
In this example, we use a data file that is very similar to the data used in the AGGREGATE
example. For more information, see the topic “Aggregating Data” on p. 65. Information was
collected for every person living in a selected sample of households. In addition to information
for each individual, each case contains a variable that identifies the household. Cases in the
same household represent related observations, not independent observations, and we want to
restructure the data file so that each group of related cases is one case in the restructured file and
new variables are created to contain the related observations.

Figure 4-12
Data file before restructuring cases to variables

The CASESTOVARS command combines the related cases and produces the new variables.

*casestovars.sps.

GET FILE = '/examples/data/casestovars.sav'.
SORT CASES BY ID_household.
CASESTOVARS

/ID = ID_household
/INDEX = ID_person
/SEPARATOR = "_"
/COUNT = famsize.
VARIABLE LABELS
Income_1 "Husband/Father Income"
Income_2 "Wife/Mother Income"
Income_3 "Other Income".

B SORT CASES sorts t